
Simulace s regresním modelem

Úloha: Simulace dat pro dal²í pouºití.

• jednorozm¥rný vstup i výstup

• obecný °ád modelu

Simulují se data z normálního, jednorozm¥rného regresního modelu

yt = a1yt−1 + a2yt−2 + · · ·+ anyt−n + b0ut + b1ut−1 + · · ·+ bnut−n + k + et

kde y, u, e jsou výstup, vstup a porucha, a, b, k jsou parametry.

P°edpoklady: e ∼ N (0, r) , r konstantní; vstup je generován p°edem.

Poznámka

Nap°. pro testování nových algoritm· je simulace dat velice d·leºitá. Data je totiº moºno simu-
lovat podle vlastních p°edstav tak, aby byly otestovány v²echny moºné varianty pouºití nového
algoritmu.

Záv¥re£né testování je ale dobré provést na reálných datech, aby algoritmus byl co nejlépe p°i-
praven k pouºití v reálném prost°edí.

Zna£ení

• y - yt,

• u - ut,

• a, b, k - a, b, k, (pouºití viz níºe),

• r - cv (pouºití viz níºe).

Volitelné parametry

• nd - po£et simulovaných dat

• Sim.Cy.ord - °ád modelu

• I_typU - volba °ízení

• Sim.Cy.tha1
Sim.Cy.thb1
Sim.Cy.thk
Sim.Cy.cv - parametry modelu
(zna£ení: Sim - simulace, Cy - komponenta pro spojitý výstup)

Doporu£ené experimenty

1. M¥¬te parametry soustavy tak, aby byla (i) pomalá, (2) rychlá.

• Rychlostí se myslí délka odezvy na jednotkový vstup.

• Tato rychlost je dána velikostí vlastních £ísel charakteristické rovnice.

• U prvního °ádu je to velmi jednoduché: pro tha blízké nule dostaneme soustavu rych-
lou, pro tha blízké jedné je soustava pomalá.

• Abychom rychlost soustavy mohli pozorovat, volíme vstup roven jedné a hodn¥ malý
rozptyl ²umu.

2. Zkuste m¥nit °ád modelu. Pozor: p°edvolby parametr· jsou nastaveny jen do °ádu 5. Potom
je pot°eba dal²í koe�cienty dodat. Nicmén¥, viditelné rozdíly jsou patrné tak do °ádu 3.

3. Zajímavé je poexperimentovat s °ádem 2. Nastavit soustavu s reálnými ko°eny, jedním
dvojnásobným ko°enem a s komplexními ko°eny. Vlastnosti jsou dány °e²ením charakte-
ristické rovnice.

Program

// Simulation of scalar regression model of order n

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

deff('ps=genpsi(t,n,y,u)','ps=[y(t-(1:n)),u(t-(0:n)) 1]','c')

nd=100; // number of data

I_typU=1; // type of input

// model

Sim.Cy.ord=2; // model order

Sim.Cy.tha=[.6 -.2 .1 -.5 .2]; // parameters at y

Sim.Cy.thb=[1 .5 -.3 -.1 .1 -.3]; // parameters at u

Sim.Cy.thk=-1; // model constant

Sim.Cy.cv=.01; // variance of noise

// model of input signal

select I_typU // choice of input

case 1, ut=rand(1,nd,'u'); // random, uniform

case 2, ut=ones(1,nd); // ones

case 3 // two steps

ut=[ones(1,fix(nd/2)), -ones(1,fix(nd/2)+1)];

case 4 // random jumps

ut(1)=1; j=1;

for i=2:nd

if rand(1,1,'u')>.85, j=rand(1,1,'n'); end

ut=[ut j];

end

case 5, ut=abs(sin(10*(1:nd)/nd)); // sinus

end

yt=zeros(1,nd);

ord=Sim.Cy.ord;

cv=Sim.Cy.cv;

th=[Sim.Cy.tha(1:ord) Sim.Cy.thb(1:(ord+1)) Sim.Cy.thk];

// regression coefficients

Sim.Cy.th=th;

// time loop

for t=(ord+1):nd // time loop

ps=genpsi(t,ord,yt,ut); // regression vector

yt(t)=ps*th'+cv*rand(1,1,'n'); // model simulation

end

Sim.Cy.yt=yt;

Sim.Cy.ut=ut;

// results

s=1:nd;

plot(s,yt(s),s,ut(s))

legend('output','input');

title('Simulation with regression model')

set(gca(),'data_bounds',[1 nd min([yt,ut])-.1 max([yt,ut])+.1])

save _data/dataT11.dat Sim

Simulace s vícerozm¥rným regresním modelem

Úloha: Simulace dat pro dal²í pouºití, zacházení s vícerozm¥rnými daty.

• dvourozm¥rný vstup i výstup (lze m¥nit)

• první °ád modelu (lze m¥nit)

Simulují se data z normálního, dvourozm¥rného regresního modelu

yt = b0ut + a1yt−1 + b1ut−1 + k + et

kde y, u, e jsou výstup, vstup a porucha, a, b, k jsou (maticové) parametry.

P°edpoklady: e ∼ N (0, r) , r konstantní; vstup je generován p°edem.

Poznámka

Obecn¥ - yt má dimenzi ny a ut má dimenzi nu. Pak ai jsou matice ny×ny , bi matice ny×nu,
k a et jsou sloupcové vektory o dimenzi ny.

Regresní vektor je

ψt =
[
u
′

t, y
′

t−1, u
′

t−1 · · · y
′

t−n, u
′

t−n, 1
]′
,

je to sloupcový vektor s dimenzí nf=(n+1)*nu+n*ny+1 a

vektor parametr· je
θ = [b0, a1, b1 · · · an, bn, k]

coº je matice s dimenzí ny×nf.

Pozor: Ve vícerozm¥rném p°ípad¥ má sou£in regresního vektoru a parametru tvar

θψt

a výsledkem je sloupec s dimenzí ny

Zna£ení

• y - yt,

• u - ut,

• a, b, k - a, b, k, (pouºití viz níºe),

• r - cv (pouºití viz níºe).

Volitelné parametry

• nd - po£et simulovaných dat

• Sim.Cy.ord - °ád modelu

• I_typU - volba °ízení

• Sim.Cy.thb0
Sim.Cy.tha1
Sim.Cy.thb1
Sim.Cy.thk - parametry modelu
(zna£ení: Sim - simulace, Cy - komponenta pro spojitý výstup)

Doporu£ené experimenty

1. M¥¬te parametry simulované soustavy a sledujte vstup a výstup soustavy na grafech.

2. M¥¬te po£et vstup· a výstup· soustavy.
Pozor: parametry soustavy jsou nastaveny pro dva vstupy a dva výstupy. Pokud chcete
tyto rozm¥ry m¥nit, musíte dodat do programu vlastní parametry. Ty pak ur£ují po£et
vstup· a výstup·. Parametry u mnohorozm¥rné soustavy jsou matice.

Program

// Simulation of a two-dimensional first order regression model

// -----------------------------------------------

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=100; // length of data

I_typU=2; // type of control (see below)

// model for simulation

Sim.Cy.ord=1; // model order

Sim.Cy.thb0=[1 .2; -.5 1]; // parameters at ut

Sim.Cy.tha1=[.4 .2; .1 .5]; // parameters at yt

Sim.Cy.thb1=[.1 -2; -.2 .1];

Sim.Cy.thk=[0; 0]; // constant (model absolute term)

Sim.Cy.th=[Sim.Cy.thb0 Sim.Cy.tha1 Sim.Cy.thb1 Sim.Cy.thk];

Sim.Cy.cv=.1; // noise variance

[my mu]=size(Sim.Cy.thb0);

yt=zeros(my,nd);

yt(1)=1; yt(2)=3; // initial conditions for output

// definition of the control variable

select I_typU

case 1, ut=rand(mu,nd); // random

case 2

z=1;

for i=1:mu

ut(i,1:nd)=z*ones(1,nd);

z=-z;

end

end

// TIME LOOP OF THE SIMULATION

ord=Sim.Cy.ord; // model order

sd=sqrt(Sim.Cy.cv);

th=Sim.Cy.th; // regression coefficients

for t=3:nd

ps=genps(ord,t,yt,ut); // regression vector

yt(:,t)=th*ps+sd*eye(my)*rand(my,1); // simulation

end

Sim.Cy.yt=yt;

Sim.Cy.ut=ut;

save _data/dataT11N.dat Sim // save data

// RESULTS OF THE SIMULATION

set(scf(1),'position',[600 100 600 600])

subplot(211),plot(ut','.'),title('Input')

subplot(212),plot(yt'),title('Output')

Simulace s kategorickým modelem

Úloha: Simulace dat pro dal²í pouºití.

• dvouhodnotové veli£iny

• model s pam¥tí a °ízením

Simulují se data z kategorického modelu, de�novaného tabulkou

[ut, zt−1] zt = 1 zt = 2
1, 1 0.9 0.1
1, 2 0.8 0.2
2, 1 0.3 0.7
2, 2 0.1 0.9

kde z, u jsou výstup a vstup, parametry jsou pravd¥podobnosti, tedy nezáporná £ísla se sou£tem
jedna v kaºdém °ádku.

P°edpoklady: parametry jsou konstantní; vstup je generován p°edem.

Poznámka

1. Neur£itost modelu je dána volbou parametr·. Jsou-li jejich hodnoty blízko 0 a 1, má v sob¥
model málo neur£itosti. Pro parametry blízké 0.5 je model velmi neur£itý.

2. Pro generování diskrétního výstupu podle pravd¥podobnostní tabulky th pouºijeme násle-
dující postup

Funkce j=psi2row(· · · ) dává °ádek j tabulky parametr· ur£ený aktuálním °ízením a minu-
lým výstupem

P°íkaz

zt(t)=sum(rand(1,1,'u')>cumsum(th(j,:)))+1;

funguje takto (nap°. pro °ádek 1 a generované náhodné £íslo 0.941)

rand(1,1,'u')>cumsum(th(j,:))

0.941>[.9, 1] = [0, 1]

dále

sum([0, 1])+1=1+1=2

Hodnota výstupu bude z(t)=2.

Zna£ení

• z - yt,

• u - ut,

• θ - th

Volitelné parametry

• nd - po£et simulovaných dat

• I_typU - volba °ízení

• Sim.Cz.th - parametry modelu
(zna£ení: Sim - simulace, Cz - komponenta pro diskrétní výstup)

Doporu£ené experimenty

1. Zvolte parametry simulované soustavy tak, aby

(a) v soustav¥ nebyla ºádná náhoda - nap°íklad tak, aby výstup neustále p°eskakoval z
jedni£ky na dvojku.

(b) v soustav¥ bylo jen málo neur£itosti - výstup se chová tak, jak ur£uje vstup a minulý
výstup a jen ob£as nastane porucha.

(c) v soustav¥ bylo hodn¥ neur£itosti - vliv vstupu a minulého výstupu se prakticky ztrácí

2. M¥¬te typ vstupu a ur£ete, kdy jsou nejlépe viditelné vlastnosti simulovaného systému
podle zadaných parametr· modelu.

Program

// Simulation of categorical model f(y(t)|u(t),y(t-1)) with y,u=1,2

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=50; // number of data

I_typU=1;

Sim.Cz.th=[.9 .1 // model parameters

.8 .2

.3 .7

.1 .9];

select I_typU // choice of input

case 1 then

ut=fix(2*rand(1,nd,'u'))+1; // random

case 2 then // two steps

ut=[ones(1,fix(nd/2)), 2*ones(1,fix(nd/2))];

case 3 // regular 1 2 1 2 1 2 ...

ut=ones(1,nd); ut(2:2:nd)=2;

end

zt=ones(1,nd);

// time loop

th=Sim.Cz.th;

for t=2:nd // time loop

j=psi2row([ut(t),zt(t-1)],[2,2]); // row in parameter table

zt(t)=sum(rand(1,1,'u')>cumsum(th(j,:)))+1;

// generation of y

end

Sim.Cz.zt=zt;

Sim.Cz.ut=ut;

// results

s=1:nd;

plot(s,zt(s),'.:',s,ut(s),'o')

legend('output','input');

title('Simulated data')

set(gca(),'data_bounds',[1 nd .9 2.1])

save _data/dataT12.dat Sim

Odhad s regresním modelem

Úloha: Odhad parametr· modelu pro dal²í pouºití.

• jednorozm¥rný vstup i výstup

• obecný °ád modelu

Jako data se vyuºívá výsledku simulace ze souboru T11SimReg1 který se ukládá jako datový
soubor dataT11.dat v adresá°i _data a odkud je nataºen. Nic ale nebrání tomu, vyrobit si vlastní
data - bu¤ p°ímo v souboru T11SimReg1 nebo si data uloºit na disk a natáhnout si je tak, jak
je to p°ipravené.

Vlastní odhad se provádí postupným p°epo£tem statistik V a κ v £asové smy£ce a teprve po
jejím ukon£ená jsou spo£teny bodové odhady parametr· modelu ve tvaru

yt = a1yt−1 + a2yt−2 + · · ·+ anyt−n + b0ut + b1ut−1 + · · ·+ bnut−n + k + et

kde y, u, e jsou výstup, vstup a porucha, a, b, k jsou parametry.

P°edpoklady: e ∼ N (0, r) , r konstantní; vstup je generován p°edem.

Poznámka

Teorie bayesovského odhadování je zaloºena na úplném popisu parametr· pomocí aposteriorní
hustoty pravd¥podobnosti. Vyuºití této informace v podob¥ celé hustoty pravd¥podobnosti je ale
zna£n¥ komplikované. Z této hustoty (tedy p°ímo z napo£ítaných statistik odhadu) je moºno
spo£ítat bodové odhady parametr·. Jejich vyuºití je jednoduché a velmi £asto posta£ující.

Zna£ení

1. y - yt

2. u - ut,

3. a, b, k - a,b,k (parametry)

4. r - cv (rozptyl ²umu)

5. thE - odhadnuté parametry

6. ordE - °ád odhadovaného modelu

Volitelné parametry

• I_dataSim - data: simulovaná, reálná,

• !!! ostatní parametry se p°ebírají ze simulace.

Doporu£ené experimenty

1. Jeden ze základních p°edpoklad· dobrého odhadu jsou data, která skute£n¥ vypovídají o
vlastnostech soustavy.
Zkuste m¥nit po£et dat, vstupující do odhadu a porovnejte kvalitu odhadu pomocí p°i-
pravené jednokrokové predikce (simulace z odhadnutého modelu).

2. Do odhadu je moºno zavést tzv. exponenciální zapomínání. To zp·sobí, ºe vliv star²ích
m¥°ených dat se potla£uje a tím se zd·razní význam aktuálních dat. To má význam,
jestliºe se vlastnosti soustavy pomalu m¥ní nebo jestliºe v minulosti byla zm¥°ena n¥která
data chybová. Historie se pomalu zapomíná. Faktor zapomínání ϕ se volí blízký jedné
(0.99, 0.995, 0.999) se do rovnic pro p°epo£et statistik zavede takto

Vt = ϕVt−1 + ΨΨ′

κt = ϕκt−1 + 1

Zkuste pouºít r·zní hodnoty zapomínání p°i odhadu a porovnejte kvalitu odhadu v jed-
notlivých p°ípadech.

3. Zkuste odhadovat s modelem r·zného °adu a sledujte vliv na kvalitu odhadu.

Program

// Estimation of scalar regression model of order n

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

deff('ps=genpsi(t,n,y,u)','ps=[y(t-(1:n)),u(t-(0:n)) 1]','c')

I_dataSim=2; // get data: 1=simulated, 2=real

ordE=2; // model order (valid for real data)

select I_dataSim

case 1

load _data/dataT11.dat Sim // load reults of task T11

yt=Sim.Cy.yt; // or use your own data yt, ut

ut=Sim.Cy.ut; // and define order of the model

ord=Sim.Cy.ord;

case 2

load _data/dataReg1.dat yt ut // from file _data/dataReg1.dat

end

nd=length(yt);

if I_dataSim==1,

ordE=Sim.Cy.ord;

th=Sim.Cy.th;

end

nPsi=2*ordE+3; // length of regression vector

V=1e-8*eye(nPsi,nPsi); // comput. of information matrix

for t=(ordE+1):nd

Ps=[yt(t) genpsi(t,ordE,yt,ut)]';

V=V+Ps*Ps';

end

thE=inv(V(2:$,2:$))*V(2:$,1); // point estimates of parameters

Est.Cy.ord=ordE;

Est.Cy.V=V;

Est.Cy.th=thE;

// results

if I_dataSim==1

disp('Simulated parmeters')

disp(th)

end

disp('Estimated parmeters')

disp(thE')

if I_dataSim==1

if length(th)==length(thE)

bar([th' thE])

title('Simulated and estimated parameters')

end

else

bar(thE)

title('Estimated parameters')

end

save _data/dataT21.dat Est

Odhad s kategorickým modelem

Úloha: Odhad parametr· modelu pro jeho dal²í pouºití.

• vstup i výstup jsou binární (mají hodnoty 1 a 2)

• model s pam¥tí a °ízením f (zt|ut, zt−1)

Jako data se vyuºívá bu¤ simulace z datového souboru dataT12.dat (generovaného souborem
T12SimCat) nebo reálných dat dataCat.dat - diskretizované hodnoty plyn a moment motoru.
Reálná rada se generují souborem generDatCat.sce, kde lze získat podrobn¥j²í informace o datech.
Oba datové soubory jsou uloºeny v adresá°i _data. Nic ale nebrání tomu, vyrobit si vlastní data
- bu¤ p°ímo v souboru T12SimCat nebo si data uloºit na disk a natáhnout si je tak, jak je to
p°ipravené.

Vlastní odhad se provádí postupným p°epo£tem statistiky V (matice 4x2), na za£átku nulová
nebo s apriorními statistikami. P°epo£et je

j=psi2row([u(t),z(t-1)],[2, 2]) - £íslo °ádku tabulky

V(j,z(t))=V(j,z(t))+1 - p°epo£et aktuálního prvku

P°edpoklady: diskrétní vstup i výstup, matice V má obecn¥ po£et °ádk· roven sou£inu po£tu
hodnot veli£in z regresního vektoru a po£et sloupc· roven po£tu hodnot výstupu.

Poznámka

Apriorní hodnoty statistiky lze ur£it p°edb¥ºným stanovením pom¥ru koe�cient· v jednotlivých
°ádcích statistiky, sílu po£áte£ní informace zadáme absolutní velikosti £ísel v apriorní statistice.

Zna£ení

1. z - zt (diskrétní výstup)

2. u - ut,

3. θ - th

4. thE - odhadnuté parametry

Volitelné parametry

• I_dataSim - data: simulovaná, reálná,

• (ostatní parametry se p°ebírají ze simulace)

Doporu£ené experimenty

1. Testujte simulovaná data pro r·zné simulované soustavy. Zkuste porovnat odhad soustavy
s hodn¥/málo neur£itosti.

2. Z reálných dat jsou k dispozici je²t¥ �spot°eba� a �rychlost�. Zkou²ejte odhadovat model
pro r·zné kombinace t¥chto veli£in v roli výstupu a vstupu. Porovnejte výsledky.

Program

// Estimation of categorical model f(y(t)|u(t),y(t-1)) with y,u=1,2

// --- on-line update of statistics, off-line point estimates

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

I_dataSim=1; // get data: 1=simulated (in T12SimCat.sce)

// 2=real (from file dataCat.dat)

if I_dataSim==1

load _data/dataT12.dat Sim // load of simulated data

zt=Sim.Cz.zt;

ut=Sim.Cz.ut;

th=Sim.Cz.th;

else

load _data/dataCat.dat sp co ga mo

zt=mo; // volba diskrétních

ut=ga; // veli£in (tady mo, ga)

end

nd=length(zt); // number of data

V=zeros(4,2); // statistics

for t=2:nd // time loop

j=psi2row([ut(t),zt(t-1)],[2,2]); // row in parameter table

V(j,zt(t))=V(j,zt(t))+1; // generation of y

end

thE=fnorm(V,2); // point estimates

Est.Cz.V=V;

Est.Cz.th=thE;

// results

if I_dataSim==1

bar([th(:,1) thE(:,1)])

title('Simulated and estimated parameters (first column of the model table)')

legend('simulated','estimated');

else

bar(thE(:,1))

title('Estimated parameters - real data (first column)')

end

save _data/dataT22.dat Est

P°edpov¥¤ výstupu s regresním modelem

Úloha: P°edpov¥¤ výstupu soustavy jako informace o soustav¥ nebo pro dal²í pouºití.

• jednorozm¥rný vstup i výstup

• obecný °ád modelu

• obecný po£et krok· predikce

Jako data se vyuºívá výsledku simulace ze souboru T12SimCat1 který se ukládá jako datový
soubor dataT11.dat v adresá°i _data a odkud je nataºen. Nic ale nebrání tomu, vyrobit si
vlastní data - bu¤ p°ímo v souboru T11SimReg1 nebo si data uloºit na disk a natáhnout si je
tak, jak je to p°ipravené. Odhad parametr· modelu se provádí v souboru T21EstReg1, a to v
batchové (jednorázové) form¥. Pro tyto bodové odhady parametr· se provádí predikce na celém
£asovém intervalu.

Pouºitý model má tvar

yt = a1yt−1 + a2yt−2 + · · ·+ anyt−n + b0ut + b1ut−1 + · · ·+ bnut−n + k + et

kde y, u, e jsou výstup, vstup a porucha, a, b, k jsou bodové odhady parametr·.

Predikce se provádí postupným dosazováním dat nebo predikcí z minulých krok· do modelu.
Tedy nap°. pro 2 °ád modelu a predikci na 2 kroky platí

ŷt = a1yt−1 + a2yt−2

ŷt+1 = a1ŷt + a2yy−1

ŷt+2 = a1ŷt+1 + a2ŷt+2

kde st°í²ka ozna£uje bodovou predikci výstupu.

P°edpoklady: e ∼ N (0, r) , r konstantní; vstup je generován p°edem.

Poznámka

Práv¥ v jednoduchosti uvaºované predikce se nejlépe ukazuje, jak je výhodné pouºití bodových
odhad· parametr· modelu soustavy. Pouºití celé aposteriorní hustoty pravd¥podobnosti by vedlo
na tak sloºitý algoritmus, ºe bychom se nevyhnuli pracnému numerickému °e²ení.

Zna£ení

• y - yt,

• u - ut,

• a, b, k - a, b, k,

• r - cv (parametry ze simulace),

• thE - odhadnuté parametry,

• ordE - °ád odhadovaného modelu,

Volitelné parametry

• np - po£et krok· predikce

Doporu£ené experimenty

1. M¥¬te délku predikce a kontrolujte její kvalitu v grafu.

2. P°idejte dal²í kriterium kvality, kterým je pr·m¥rná kvadratická chyba predikce.
Chybu predikce ur£íte jako rozdíl mezi hodnotami výstupu a jejich predikcí

epi = yi − ypi, i = 1, 2, · · · , nd

Kriterium potom je

SE =
1

nd

nd∑
i=1

ep2i

Pr·m¥r chyby predikce by m¥l být blízko nuly, kvadratická chyba SE by m¥la být co
nejmen²í.

3. M¥¬te parametry simulované soustavy (zejména rozptyl ²umu) a sledujte kvalitu predikce.
Soustavu lze m¥nit v simulaci, tj. ve souboru T11SimReg1.sce.

4. Podstatný vliv na kvalitu odhadu a tedy i predikce má volba vstupního signálu. Jestliºe
vstup je p°íli² hladký bez skok·, soustava není pro odhad dostate£n¥ vybuzena, parametry
se ur£í nedostate£n¥ p°esn¥ a predikce není dobrá.

Program

// Prediction np-steps with scalar regression model of order n

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

// definition of regression vector

deff('ps=genpsi(t,n,y,u)','ps=[y(t-(1:n)),u(t-(0:n)) 1]','c')

load _data/dataT11.dat Sim // results from simulation T11SimReg1.sce

yt=Sim.Cy.yt;

ut=Sim.Cy.ut;

th=Sim.Cy.th;

cv=Sim.Cy.cv;

ord=Sim.Cy.ord;

load _data/dataT21.dat Est // results from estimation T21EstReg1.sce

thE=Est.Cy.th;

ordE=Est.Cy.ord;

np=5; // no of prediction steps (1=prediction from the model)

nd=length(yt);

yp=zeros(1,nd);

for t=(ordE+1):(nd-np+1) // time loop

yi=yt(1:(t-1)); // old data (at time t)

for j=0:(np-1) // loop of prediction

ps=genpsi(t+j,ordE,yi,ut);// construction of regression vector

yi(t+j)=ps*thE; // auxiliary prediction

end

yp(t+np-1)=yi(t+np-1); // final prediction at t

end

Pre.Cy.np=np;

Pre.Cy.yp=yp;

// Results

s=(ordE+1):(nd-np);

plot(s,yt(s),s,yp(s))

title('Output and its prediction')

legend('output','prediction');

save _data/dataT31.dat Sim Est Pre

P°edpov¥¤ výstupu s kategorickým modelem

Úloha: P°edpov¥¤ výstupu jako informace o soustav¥ nebo pro dal²í pouºití.

• vstup i výstup s hodnotami 1, 2

• model s pam¥tí a °ízením

• jednokroková a n-kroková predikce

Jako data se vyuºívá výsledku simulace ze souboru T12SimCat který se ukládá jako datový
soubor dataT12.dat v adresá°i _data a odkud je nataºen. Nic ale nebrání tomu, vyrobit si
vlastní data - bu¤ p°ímo v souboru T12SimCat nebo si data uloºit na disk a natáhnout si je
tak, jak je to p°ipravené. Odhad parametr· modelu se provádí v souboru T22EstCatX v první
variant¥, tj. pro I_dataSim=1a jeho výsledky jsou uloºeny v datovém souboru dataT22.dat.

Pouºitý model má tvar tabulky

[ut, yt−1] yt = 1 yt = 2
1, 1 Θ1|11 Θ2|11
1, 2 Θ1|12 Θ2|12
2, 1 Θ1|21 Θ2|21
2, 2 Θ1|22 Θ2|22

kde y, u jsou výstup a vstup, Θi|jk jsou pravd¥podobnosti.

K dispozici jsou 2 programy:

1. T32PreCatN.sce - jednokroková predikce (simulace z odhadnutého modelu).

2. T32PreCatN.sce - vícekroková predikce pro np krok· dop°edu.

Predikce v T32CatN se provádí postupným dosazováním dat nebo bodových predikcí z minulých
krok· do modelu.

P°edpoklady: Dvouhodnotová data

Zna£ení

• y - yt,

• u - ut,

• Θ̂t - thE - odhadnuté parametry.

Volitelné parametry

• np - po£et krok· v predikci.

Doporu£ené experimenty

1. Volte r·zné druhy soustavy od málo neur£itých aº po deterministické. Volbu je t°eba
provést v simulaci, tj v souboru T12SimCat.

2. Zkuste r·zný po£et krok· predikce np a porovnejte výsledky. Najd¥te hranici pro délku
predikce, kdy je²t¥ dostáváme p°ijatelné výsledky. Tato hranice je samoz°ejm¥ pro kaºdý
typ simulované soustavy jiná.

3. Upravte program tak, aby predikce pracovala se skute£nými parametry simulace (známé
parametry soustavy). Porovnejte výsledky predikce se známými a odhadnutými parametry.

Program

Jednokroková predikce s bodovými odhady

// Prediction one-step ahead with scalar categorical model

// - model is binary + memmory + control f(z(t)|u(t),z(t-1))

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

load _data/dataT12.dat Sim // results from task T12SimCat.sce

zt=Sim.Cz.zt; // the data zt, ut

ut=Sim.Cz.ut; // and parameters th are

th=Sim.Cz.th; // extracted from simulation

load _data/dataT22.dat Est // results from estimation are taken

thE=Est.Cz.th; // from task T22EstCat.sce

nd=min(50,length(zt)); // number of data (minimum is 50)

yp=ones(1,nd);

for t=2:nd // time loop

j=psi2row([ut(t),zt(t-1)],[2,2]); // row in the table

yp(t)=sum(rand(1,1,'u')>cumsum(thE(j,:)))+1; // prediction at t

end

Pre.Cz.np=1; // only 1 step prediction

Pre.Cz.yp=yp;

// Results

s=1:nd;

printf('There is %d wrong predictions from %d\n',sum(zt(s)~=yp(s)),length(s))

plot(s,zt(s),'b:o',s,yp(s),'r:.')

legend('output','prediction');

title 'Output and its prediction'

save _data/dataT32.dat Sim Est Pre

Vícekroková predikce s bodovými odhady

// Prediction np-step with categorical model

// - the old outputs are substituted by point estimates

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

load _data/dataT12.dat Sim // results from task T12SimCat.sce

zt=Sim.Cz.zt; // from simulation, the data

ut=Sim.Cz.ut; // zt, ut and parameters th

th=Sim.Cz.th; // are extracted

load _data/dataT22.dat Est // results from task T22EstCat.sce

thE=Est.Cz.th;

np=3; // numb. of prediction steps (1=prediction from the model)

nd=length(zt);

yp=zeros(1,nd);

for t=2:(nd-np+1) // time loop

yi=zt(1:(t-1)); // old data (at time t)

for k=0:(np-1) // loop of prediction

j=psi2row([ut(t+k),yi(t+k-1)],[2,2]); // row of the table

yi(t+k)=sum(rand(1,1,'u')>cumsum(thE(j,:)))+1; // point prediction

end

yp(t+np-1)=yi(t+np-1); // final prediction at time t

end

Pre.Cz.np=np;

Pre.Cz.yp=yp;

// results

s=(fix(nd/2)):(nd-np);

printf(' %d wrong predictions from %d\n',sum(zt(s)~=yp(s)),length(s))

plot(s,zt(s),'b:o',s,yp(s),'r:.')

legend('output','prediction');

save _data/dataT32.dat Sim Est Pre

P°edpov¥¤ výstupu s kategorickým modelem

Úloha: P°edpov¥¤ výstupu jako informace o soustav¥ nebo pro dal²í pouºití.

• £tvercová tabulka modelu, tj. model f (yt|yt−1Θ)

• obecná n-kroková predikce

• predikce s bodovými odhady ve tvaru maximum pravd¥podobnosti nebo st°ední hodnota

• volba, zda predikce b¥ºí se známými nebo odhadovanými parametry

Data jsou v programu simulována.

Pouºitý model má tvar tabulky

[ut, yt−1] yt = 1 yt = 2
1, 1 Θ1|11 Θ2|11
1, 2 Θ1|12 Θ2|12
2, 1 Θ1|21 Θ2|21
2, 2 Θ1|22 Θ2|22

kde y, u jsou výstup a vstup, Θi|jk jsou pravd¥podobnosti.

Predikce se provádí mocn¥ním tabulky modelu (jako £tvercové matice) a násobením po£áte£ní
hodnotou.

P°edpoklady: Model s £tvercovou tabulkou.

Zna£ení

• y - yt,

• u - ut,

• Θ̂t - thE - odhadnuté parametry.

Volitelné parametry

• np - po£et krok· v predikci

• I_est - volba predikce se známými nebo odhadovanými parametry

• I_pred - volba bodových odhad· (maximum pravd¥podobnosti nebo st°ední hodnota)

Doporu£ené experimenty

1. Volte r·zné druhy soustavy od málo neur£itých aº po deterministické.

2. Zkuste r·zný po£et krok· predikce np a porovnejte výsledky. Najd¥te hranici pro délku
predikce, kdy je²t¥ dostáváme p°ijatelné výsledky. Tato hranice je samoz°ejm¥ pro kaºdý
typ simulované soustavy jiná.

3. Porovnejte výsledky predikce se známými a odhadovanými parametry.

4. Upravte program tak, aby pracoval s o�-line odhadnutými parametry. Porovnejte výsledky
se známými, odhadovanými a odhadnutými parametry. Navrhn¥te kriterium pro porov-
nání.

Program

// Prediction with a "square" discrete model f(z(t)|z(t-1),th)

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

np=5; // prediction length

nd=500; // data sample length

I_est=2; // estimation: 1=yes, 2=no

I_pred=2; // prediction: 1=arg max, 2=expect.

T=[.99 .005 .005; // simulation model parameter

.01 .98 .01

.02 .01 .97];

// SIMULATION

z(1)=1; // initial output

for t=2:nd // data generation

z(t)=(sum(rand(1,1,'unif')>(cumsum(T(z(t-1),:)))))+1;

end

// ESTIMATION AND PREDICTION

S=ones(T); // initial model statistics

vals=(1:max(size(T)))'; // values of output

for t=2:nd

S(z(t-1),z(t))=S(z(t-1),z(t))+1; // statistics update

// known or estimated parameters

select I_est // estimation 1 = no, 2 = yes

case 1 then

Te=T;

case 2 then

Te=fnorm(S,2); // parameter point estimates

end

Tp=Te^np; // model for prediction

// type of prediction point estimates

select I_pred

case 1 then

[xxx yy]=max(Tp(z(t),:));

yp(t+np)=yy;

case 2 then

yp(t+np)=Tp(z(t),:)*vals;

end

end

// RESULTS

s=2:nd;

set(scf(1),'position',[100 100 500 400])

plot(s,z(s),'o',s,yp(s),'.')

set(gca(),'data_bounds',[1 nd .9 3.1]);

title 'Discrete data and their prediction'

legend('data','prediction');

Odhad stavu se stavovým modelem

• stavový lineární model se známými parametry

• simulovaná data

V programu se provádí simulace se stavovým modelem a g generovaného vstupu a výstupu se
pomocí Kalmanova �ltru pr·b¥ºn¥ odhaduje stav. Kalman·v �ltr je realizován procedurou

[x, xf, rx, yp]=Kalman(x, yt, ut, M, N, F, A, B, G, rw, rv, rx)

kde x je p°epo£ítávaný stav, xf je výsledný odhad stavu v daném £ase, rx je p°epo£ítávaná
kovariance stavu, yp je predikce výstupu, yt a ut jsou data, M, N, F, A, B, G jsou parametry
stavového modelu, rw a rv jsou kovariance stavu a ²umu ve stavovém modelu.

Pouºitý model má tvar

xt+1 = Mxt +Nut + F + wt

yt = Axt +But +G+ vt

kde xt je v programu ozna£en jako xf

P°edpoklady: Známe parametry modelu, linearita modelu

Sci zna£ení: xt−1/xt/xt+1 - x, xt - xf, (ostatní viz Kalman·v �ltr naho°e)

Úloha: Odhad neznámé veli£iny, �ltrace za²um¥ného signálu.

Poznámka

Pro správný start odhadování je d·leºité správné nastavení kovarian£ních matic. Matice rw se
nastavuje v¥t²inou jako diagonální s velkými £ísly na diagonále (103−105). Kovarian£ní matice
rw a rv by m¥ly odpovídat kovarian£ním maticím ²um· wt a vt z modelu. POZOR: nejsou to
kovarian£ní matice stavu a výstupu ale jejich poruch. Na správném nastavení t¥chto matic velmi
záleºí celý odhad.

Doporu£ené experimenty

1. M¥¬te parametry stavového modelu pro simulaci a sledujte, kdy je odhad stavu obtíºný.
Poznámka
Jestliºe matice

[
A, AM, AM2, · · · , AMn−1] , kde n je dimenze stavu, má hodnost men²í

neº n, pak systém je tzv. nepozorovatelný a stav nelze odhadnout.

2. M¥¬te rozptyly ²um· rw a rv. Pro velké rozptyly bude odhad mén¥ p°esný.

3. M¥¬te po£áte£ní kovarian£ní matici rx. Nastavená hodnota 1000 na diagonále znamená, ºe
Kalmanovu �ltru necháváme volnost v jeho odhadu z dat. Pro men²í diagonálu odhad tzv.
�utahujeme�, tj. dáváme v¥t²í váhu po£áte£ním podmínkám. Tím se ale odhad stabilizuje -
sniºuje se nebezpe£í, ºe n¥jaká hodn¥ chybná data na za£átku celý odhad rozhodí natolik,
ºe nekonverguje.

Program

// Estimation of state variable with KF

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=100; // number of data

// State-space model

Fil.Cy.rw=.1*eye(2); // state covariance

Fil.Cy.rv=.01; // output covariance

Fil.Cy.rx=1000*eye(2); // state estimate covariance

Fil.Cy.M=[.2 .1; // state-space

.2 .4]; // model

Fil.Cy.N=[1; -1]; // parameters

Fil.Cy.F=[1; -1]; // "

Fil.Cy.A=[1 0]; // "

Fil.Cy.B=1; // "

Fil.Cy.G=2; // parameters

// declarations

xt=zeros(2,nd); // state declaration (initial state)

x=zeros(2,nd); // initial state estimate

ut=zeros(1,nd); // input

yt=zeros(1,nd); // output

yp=zeros(1,nd); // output prediction

rw=Fil.Cy.rw; rv=Fil.Cy.rv; rx=Fil.Cy.rx; // KF covariances

M=Fil.Cy.M; N=Fil.Cy.N; F=Fil.Cy.F; // state model parameters

A=Fil.Cy.A; B=Fil.Cy.B; G=Fil.Cy.G; // output model parameters

// TIME LOOP

for t=2:nd

// simulation

xt(:,t) = M*xt(:,t-1)+N*ut(t)+F+rw*rand(2,1,'norm');

yt(t) = A*xt(:,t) + B*ut(t)+G+rv*rand(1,1,'norm');

// estimation

[x(:,t),rx,yp(t)]=...

Kalman(x(:,t-1),yt(t),ut(t),M,N,F,A,B,G,rw,rv,rx);

end

Fil.Cy.ut=ut;

Fil.Cy.yt=yt;

Fil.Cy.xt=x;

// RESULTS

s=1:nd;

subplot(311),plot(s,xt(1,s),s,x(1,s))

set(gca(),"data_bounds",[1,nd,0 2]);

title('First state and its estimate')

subplot(312),plot(s,xt(2,s),s,x(2,s))

set(gca(),"data_bounds",[1,nd,-2 1]);

title('Second state and its estimate')

subplot(313),plot(s,yp(s),s,yt(s))

set(gca(),"data_bounds",[1,nd,2 4]);

title('Output and its prediction')

set(gcf(),"position",[500 50 600 500])

// Note: xf is the result of state estimation (not x)

// at time t, xf(t) is the result. x(t+1) aims at the next time

save _data/dataT44.dat Fil

Odhad stavu s nelineárním stavovým modelem

• stavový model se známými parametry

• nelineární model

• simulovaná data

V programu se provádí simulace s nelineárním stavovým modelem.

Pro odhad stavu je tento model linearizován pomocí prvých dvou £len· Taylorova rozvoje.
Linearizovaný model je vyuºit pro odhad stavu z generovaného vstupu a výstupu. Odhad se
provádí pomocí Kalmanova �ltru. Kalman·v �ltr je realizován procedurou

[x, xf, rx, yp]=Kalman(x, yt, ut, M, N, F, A, B, G, rw, rv, rx)

kde x je p°epo£ítávaný stav, xf je výsledný odhad stavu v daném £ase, rx je p°epo£ítávaná
kovariance stavu, yp je predikce výstupu, yt a ut jsou data, M, N, F, A, B, G jsou parametry
stavového modelu, rw a rv jsou kovariance stavu a ²umu ve stavovém modelu.

Pouºitý model má tvar

x1;t+1 = x1;tx2;t − ut + w1;t

x1;t+1 = 0.5x1;t + 0.8x2;t + ut + w2;t

yt = x1;t

kde xt je v programu ozna£en jako xf

P°edpoklady: Známe parametry modelu, linearizace modelu pomocí Taylorova rozvoje

Sci zna£ení: xt−1/xt/xt+1 - x, xt - xf, (ostatní viz Kalman·v �ltr naho°e)

Úloha: Odhad neznámé veli£iny, �ltrace za²um¥ného signálu.

Poznámka

Pro správný start odhadování je d·leºité správné nastavení kovarian£ních matic. Matice rw se
nastavuje v¥t²inou jako diagonální s velkými £ísly na diagonále (103−105). Kovarian£ní matice
rw a rv by m¥ly odpovídat kovarian£ním maticím ²um· wt a vt z modelu. POZOR: nejsou to
kovarian£ní matice stavu a výstupu ale jejich poruch. Na správném nastavení t¥chto matic velmi
záleºí celý odhad.

Doporu£ené experimenty

1. M¥¬te rozptyly ²um· rw a rv a sledujte jejich efekt na kvalitu odhadu.

2. Zkuste zm¥nit simulovanou soustavu. Ale pozor! Pro novou soustavu musíte p°epo£ítat
matice M, N, A, B, F a G které plynou z linearizace nelineárního modelu.

Program

// State estimation (nonlinear model)

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=200; // length of data

ut=.1*(1-rand(1,nd,'norm')); // control

x1=0; x2=-1; // initial state

xs=zeros(2,nd); xs(:,1)=[x1;x2];

yt=zeros(1,nd);

// SIMULATION

for t=2:nd

x1=x1*x2+.1*x2-ut(t)+.01*rand(1,1,'norm'); // state

x2=.5*x1+.8*x2+ut(t)+.01*rand(1,1,'norm'); // equations

yt(t)=x2; // output equation

xs(:,t)=[x1;x2]; // store the actual sate variable

end

xx=[0;0]; // initial state estimate

rw=.1*eye(2,2); rv=.1; rx=1e3*eye(2,2); // model and initial covariances

xt=zeros(2,nd); rr=zeros(2,nd); rr(:,1)=diag(rx);

// ESTIMATION

for t=2:nd

// parameters of the linearized model

M=[xx(2) xx(1)+.1; .5 .8];

N=[-1;1];

F=[-.1*xx(1)+(.1-xx(2))*xx(2);0];

A=[0 1];

B=0;

G=0;

// Kalman filter

[xx,rx,yp]=Kalman(xx,yt(t),ut(t),M,N,F,A,B,G,rw,rv,rx);

xt(:,t)=xx; // stor actual state estimate

rr(:,t)=diag(rx); // stor variance of noise estimate

end

// RESULTS

// figure 1

plot(xs')

plot(xt',':')

legend('state 1','state 2','estimate 1','estimate 2',4);

title('State and its estimate','fontsize',5,'FontName','Times')

xlabel('time','fontsize',4,'FontName','Times')

ylabel('values','fontsize',4,'FontName','Times')

set(gcf(),'position',[300 100 500 400])

Odhad stavu s neznámými parametry

• stavový model

• neznámé parametry

• linearizace vzniklého nelineárního modelu

V programu se provádí simulace s nelineárním stavovým modelem.

Pro odhad stavu se stav modelu se roz²í°í o neznámé parametry a tím se stane nelineárním.
Model je dále linearizován pomocí prvých dvou £len· Taylorova rozvoje. Linearizovaný model je
vyuºit pro odhad stavu z generovaného vstupu a výstupu. Odhad se provádí pomocí Kalmanova
�ltru. Kalman·v �ltr je realizován procedurou

[x, xf, rx, yp]=Kalman(x, yt, ut, M, N, F, A, B, G, rw, rv, rx)

kde x je p°epo£ítávaný stav, xf je výsledný odhad stavu v daném £ase, rx je p°epo£ítávaná
kovariance stavu, yp je predikce výstupu, yt a ut jsou data, M, N, F, A, B, G jsou parametry
stavového modelu, rw a rv jsou kovariance stavu a ²umu ve stavovém modelu.

Pouºitý model má tvar

a=.6; // unknown model parameter

sM=[a .5 // simulation parametrs

.1 .8];

sN=[.5; .4];

sA=[1 0];

yt(:,t)=sA*x+sqrt(rv)*rand(1,1,'norm'); x=sM*x+sN*ut(:,t)+sqrt(rw)*rand(2,1,'norm');

xt+1 = Mxt +Nu+ wt

yt = Axt + vt

kde M =

[
a 0.5
1 0

]
,

[
0.5
0.4

]
, A = [1, 0] . V simulaci je a = 0.6 v odhadu je a neznámý

parametr.

P°edpoklady: Neznáme parametry modelu, linearizace modelu pomocí Taylorova rozvoje

Sci zna£ení: xt−1/xt/xt+1 - x, xt - xf, (ostatní viz Kalman·v �ltr naho°e) Parametry modelu
pro simulaci jsou sM, sN a sA.

Úloha: Odhad neznámé veli£iny, �ltrace za²um¥ného signálu.

Poznámka

Pro správný start odhadování je d·leºité správné nastavení kovarian£ních matic. Matice rw se
nastavuje v¥t²inou jako diagonální s velkými £ísly na diagonále (103−105). Kovarian£ní matice
rw a rv by m¥ly odpovídat kovarian£ním maticím ²um· wt a vt z modelu. POZOR: nejsou to
kovarian£ní matice stavu a výstupu ale jejich poruch. Na správném nastavení t¥chto matic velmi
záleºí celý odhad.

Doporu£ené experimenty

1. Stav je v tomto p°ípad¥ sloºen z p·vodních stavových prom¥nných (které se p°irozen¥ v
£ase m¥ní) a neznámých parametr· (které jsou konstantní). Vzhledem k tomu je t°eba
volit rozptyly p·vodních stav· v¥t²í, zatímco rozptyly neznámých parametr· by m¥ly být
velmi malé. Zkuste experimentáln¥.

2. Nejzajímav¥j²í je m¥nit simulovanou soustavu (v£etn¥ dimenze stavu a výstupu). Tato
zm¥na ale vyºaduje novou konstrukci roz²í°eného stavového modelu (stav je roz²í°en o
neznámé parametry) a jeho linearizace.

Program

// State estimation (model with unknown parameter)

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=100; // length of data

a=.6; // unknown model parameter

sM=[a .5 // simulation parametrs

.1 .8];

sN=[.5; .4];

sA=[1 0];

rv=.01; rw=.01*eye(2,2); // noise variances

ut=rand(1,nd,'norm');

yt=zeros(1,nd);

xx=zeros(2,1); // initial state

xt=xx;

// SIMULATION

for t=1:nd

xx =sM*xx+sN*ut(:,t)+sqrt(rw)*rand(2,1,'norm');

yt(:,t)=sA*xx+sqrt(rv)*rand(1,1,'norm');

xt(:,t)=xx;

end

xt(3,:)=a; // true parameter

Rv=.01; Rw=.01; // model covariance matrices

Rx=1e6*eye(3,3); // covariance matrix of state estimate

N=[.5; .4; 0]; // parameters of model extended by unknown

A=[1 0 0]; // parmeter

G=0;

B=0;

x=zeros(3,1);

xe=zeros(3,nd);

// ESTIMATION

for t=1:nd

M=[x(3) .5 0 // parameters of thr linearized model

.1 .8 0

0 0 1];

dM=[x(3) .5 x(1)

.1 .8 0

0 0 1];

F=(M-dM)*x;

// KALMAN FILTER

[x,Rx,yp]=Kalman(x,yt(t),ut(:,t),dM,N,F,A,B,G,Rw,Rv,Rx);

xe(:,t)=x;

end

// RESULTS

set(scf(1),'position',[100 50 1200 400])

// figure 1

subplot(131)

plot(xt(1,:),'b','linewidth',2)

plot(xe(1,:),'r','linewidth',2)

legend('state','estimate');

title('First state and its estimate','fontsize',5,'FontName','Times')

xlabel('time','fontsize',4,'FontName','Times')

ylabel('values','fontsize',4,'FontName','Times')

// figure 2

subplot(132)

plot(xt(2,:),'b','linewidth',2)

plot(xe(2,:),'r','linewidth',2)

legend('state','estimate');

title('Second state and its estimate','fontsize',5,'FontName','Times')

xlabel('time','fontsize',4,'FontName','Times')

ylabel('values','fontsize',4,'FontName','Times')

// figure 3

subplot(133)

plot(xt(3,:),'b','linewidth',2)

plot(xe(3,:),'r','linewidth',2)

legend('true parameter','estimate');

title('Estimate of the parameter','fontsize',5,'FontName','Times')

xlabel('time','fontsize',4,'FontName','Times')

ylabel('values','fontsize',4,'FontName','Times')

32



Klasi�kace s regresním modelem

• simulovaná data

• klasi�kace - t°íd¥ní dat do t°íd

• logistický model a jeho ML odhad

V programu se provádí odhad modelu logistické regrese (fáze u£ení) a pro odhadnutý model,
tj. model s pevnými parametry, se provádí t°íd¥ní dat (fáze testování). Data se za°adí do t°ídy,
která je indikována hodnotou odhadnutého výstupu (blíºe viz poznámka).

Simulace se provádí tak, ze se generují náhodné vektory x1 a x2. Vytvo°í se jejich lineární
kombinace β0 + β1x1 + β2x2 + ε (ε je ²um) a pro výsledek v¥t²í neº 0.5 se p°i°adí y = 1 jinak
y = 0.

Model v úloze u£ení testování má standardní tvar logistické regrese

zt = xtb

P (yt = 1|xt) =
exp {zt}

1 + exp {zt}

ŷt =

{
1 pro P (yt = 1|xt) > 0.5

0 pro P (yt = 1|xt) < 0.5

P°edpoklady: Dvouhodnotové y ∈ {0, 1}

Sci zna£ení: yL, xL - data pro u£ení; xT - data pro testování; c - odhad výstupu (t°ídy); yT -
správná t°ída ze simulace.

Úloha: Odhad modelu logistické regrese, odhad t°ídy dat pro m¥°ené datové záznamy (klasi�-
kace).

Poznámka

V logistické regresi se pouºívají veli£iny y - výstup a x - regresní vektor. V klasi�kaci mají jinou
interpretaci: x jsou datové záznamy, které chceme t°ídit, y je ozna£ení t°ídy, do které data pat°í.

Doporu£ené experimenty

1. Zkuste m¥nit velikost datového vzorku pro u£ení (pomocí parametru nL). Sledujte vliv
na kvalitu klasi�kace. Ta se posuzuje jako po£et ²patných klasi�kací vzhledem k po£tu
provedených klasi�kací.

2. Simulace je v programu provedena heuristicky. Nejd°íve se generují hodnoty nezávislých
prom¥nných x. Potom se provede jejich lineární kombinace a výstup y se ur£í jako 1 pro
její kladné hodnoty, v opa£ném p°ípad¥ je 0. Zkuste navrhnout svou vlastní simulaci.

3. Zkuste provést simulaci generovanou logistickým modelem, pro který si zadáte své vlastní
parametry. Porovnejte simulované a odhadnuté parametry.

Program

// Classification with logistic regression

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nL=100; // number of data for learning

nT=100; // number of data for testing

sd=1; // magnitude of disturbance

x1=rand(1,nL+nT,'norm'); // 1. regression vector

x2=rand(1,nL+nT,'unif')<.3; // 2. regression vector

x=[x1;x2]'; // full regression vector

sL=1:nL; // inteval for learning

sT=1:nT; // interval for testing

yy=2*x1-3*x2+1+sd*rand(1,nL+nT,'norm');

y=double(yy>0); // output (true classes)

yL=y(sL);

xL=x(sL,:);

// Learning

// py probabilities of y=0 and y=1

// yp prediction of y yp=py(:,2)

// yr point prediction of y yr=round(yp)

// b model parameters

[py,yp,yr,b]=lrEst([],yL,xL); // log. reg. estimation

// Testing

ct=zeros(sT);

yT=y(1,nL+sT); // true classes

xT=[ones(nT,1) x(nL+sT,:)]; // data for testing

for t=sT

z=xT(t,:)*b; // regression

yp=exp(z)/(1+exp(z)); // logistic function

ct(t)=round(yp); // class point estimates

end

// RESULTS

s=1:nT;

set(scf(1),'position',[600 50 600 600])

plot(s,yT(s),'bo',s,ct(s),'r.')

set(gca(),"data_bounds",[min(s)-.1 max(s)+.1 -.1 1.1])

title('Logistic regression')

legend('output','point estimate');

disp('Logistic regression:')

printf(' Wrong %d from %d\n',sum(yT~=ct),nT)

Adaptivní °ízení s regresním modelem

• regresní model s vektorovým vstupem a výstupem

• model obecného °ádu s neznámými parametry

• separace identi�kace a °ízení

• realizace adaptivního °ízení p°i ustupujícím horizontu

V programu se nejprve provede p°edb¥ºná identi�kace modelu z apriorních dat a spo£tou bodové
odhady parametr· modelu.

potom, v £asové smy£ce, se regresní model p°evede do stavového tvaru. Provádí optimalizace, tj.
výpo£et °ídícího p°edpisu na celém intervalu °ízení, a to proti sm¥ru £asu. Nakonec se realizuje
aplikace posledního (tedy aktuálního) p°edpisu pro °ízení. Po jeho aplikaci se generuje (zm¥°í)
nový výstup a s novými daty se p°epo£tou bodové odhady parametr·.

Pouºitý regresní model

yt = b0ut + a1yt−1 + b1ut−1 + · · ·+ anyt−n + bnut−n + k + et.

Model se ur£í zadáním parametr· u y, u a konstanty.

P°edpoklady: Konstantní parametry modelu pro simulaci.

Sci zna£ení: M, N - parametry stavového modelu, Om - penalizace.

Úloha: �ízení s mnohorozm¥rným regresním modelem.

Poznámka

Realizuje se kvadraticky optimální °ízení které má sledovat p°edem daný pr·b¥h. Tomuto pr·b¥hu
se °íká ºádaná hodnota nebo set-point.

P°i odhadu b¥hem °ízení mohou nastat potíºe s vybuzením soustavy.

Doporu£ené experimenty

1. Adaptivní °ízení ve velice sloºitá úloha, která vyºaduje ur£itou inicializaci (p°edb¥ºný
odhad parametr·) tak, aby °ízení jiº od za£átku fungovalo alespo¬ trochu �rozumn¥�.
Pokud tomu tak není, m·ºe se celá uzav°ená smy£ka tak vybudit (v¥t²inou se generují
obrovská £ísla, ta se v po£íta£i zaokrouhlí a tím se naru²í konzistence systému), ºe se
°ízení rozpadne a nekonverguje.

(a) Zkuste m¥nit délku po£áte£ního odhadu ni a sledujte kvalitu °ídícího procesu (zejména
na za£átku °ízení).

(b) Dále lze experimentovat s po£áte£ním nastavením informa£ní matice V (tady na-
stavena jako diagonální s diagonálou 0.1) a statistikou ka (£ím je v¥t²í, tím více se
prosazují po£áte£ní hodnoty parametr·).
Pozor: Matice V musí být symetrická (viz její p°epo£et).

2. Pro implementaci algoritmu adaptivního °ízení se vyuºívá metoda ustupujícího horizontu.
Délka horizontu se zadává jakoCon.Cy.nh a je p°edvolena na hodnotu 3. M¥la by být tak
velká, aby se po£ítaný °ídící zákon sta£il ustálit.
Zkuste m¥nit její hodnotu a sledujte kvalitu °ízení.

3. V kriteriu °ízení se penalizuje kvadrát výstupu a kvadrát p°ír·stku °ídící veli£iny, a to s
vahou la (ur£uje významnost penalizace °ízení proti penalizaci výstupu).

(a) Zkuste m¥nit hodnotu la ∈ (0, 1) a ud¥lejte záv¥ry o vlivu na kvalitu °ízení.

(b) Zkuste rozmyslet, jak je t°eba zm¥nit penaliza£ní matici Om, aby se penalizovaly celé
hodnoty °ízení, nikoliv p°ír·stky. Porovnejte °ízení s penalizací celé °ídící veli£iny a
jejích p°ír·stk·.

4. Pomocí parametru Iadapt je moºno p°epínat °ízení adaptivní a °ízení se známými para-
metry. Testujte a porovnejte.

Program

// ADAPTIVE CONTROL WITH DYNAMIC REG. MODEL WITH CONSTANT

// multivariate input and output

// control with setpoint

// penalization of input increments

// preliminary (prior) estimation

// ------------------------------------------------

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

ni=50; // preliminary estimation

nd=150; // control

I_adapt=0; // adaptive control 1=yes, 0=no,

Sim.Cy.ord=2; // model order, else fix prior est.

Sim.Cy.th(1).a=[.3 -.2; -.05 -.3];

Sim.Cy.th(2).a=[.1 -.1; .02 -.1];

Sim.Cy.thb0=[1 -.4; 1.3 -.4];

Sim.Cy.th(1).b=[.1 0;0 .2];

Sim.Cy.th(2).b=[.1 0;0 .1];

Sim.Cy.thk=[1; -1];

Sim.Cy.sd=.5; // noise stdev

ord=Sim.Cy.ord;

ny=size(Sim.Cy.th(1).a,1); // dimension of y

nu=size(Sim.Cy.th(1).b,2)/(1+ord); // and u

th=Sim.Cy.thb0;

for i=1:ord // parameters -> vector th

th=[th Sim.Cy.th(i).a Sim.Cy.th(i).b];

end

th=[th Sim.Cy.thk];

k=Sim.Cy.thk;

sd=Sim.Cy.sd;

Con.Cy.nh=3; // length of control interval

Con.Cy.om=1; // penalty: y(t)^2

Con.Cy.la=.01; // penalty: (u(t)-u(t-1))^2

nh=Con.Cy.nh;

[XX,XX,XX,nx,ny,nu]=reg2st(th,ord);

// penalization matrices

Om=zeros(nx,nx);

Om(1:ny,1:ny)=Con.Cy.om*eye(ny,ny); // output

nn=ny+nu;

for i=1:nu // input increment

Om((ny+i),(ny+i))=Con.Cy.la; // (only for ord>1 !!)

Om((ny+i+nn),(ny+i+nn))=Con.Cy.la;

Om((ny+i),(ny+i+nn))=-Con.Cy.la;

Om((ny+i+nn),(ny+i))=-Con.Cy.la;

end

// setpoint

g=genstp(ny,nd,[3;5],[.95;.92],[-5;10]);

// Prior estimation

yi=zeros(ny,ni);

ui=rand(nu,ni,'norm');

V=.01*eye(nx+ny+nu,nx+ny+nu);

for i=(ord+1):ni

ps=genps(ord,i,yi,ui);

yi(:,i)=th*ps + .1*rand(ny,1,'norm');

Ps=[yi(:,i); ps];

V=V+Ps*Ps';

end

// TIME LOOP

S=list();

y=zeros(ny,nd); y(:,1:2)=[10 10;-10 -10];

u=zeros(nu,nd);

R=.00001*eye(nx,nx); // regularizace

for t=(ord+1):(nd-nh)

// Estimation

if I_adapt==1

ps=genps(ord,t,y,u);

Ps=[y(:,t); ps];

V=V+Ps*Ps';

theta=v2thN(V/t,ny);

else

theta=th';

end

[M,N,A,nx,ny,nu]=reg2st(theta',ord);

// Generation of control law

for i=nh:-1:1

M(1:ny,$)=-g(:,t+i)+k;

T=R+Om;

A=N'*T*N;

B=N'*T*M;

C=M'*T*M;

S(t)=inv(A)*B;

R=C-S(t)'*A*S(t);

end

// Generation of optimal control

x=genph(ord,t,y,u);

u(:,t)=-S(t)*x;

y(:,t)=th*[u(:,t); x]+.01*rand(ny,1,'norm');

end

// Results

s=1:(nd-nh);

subplot(211)

plot(s,y(1,s),s,u(1,s),'--',s,g(1,s),'m.','markersize',2)

legend('output','input','setpoint',4);

subplot(212)

plot(s,y(2,s),s,u(2,s),'--',s,g(2,s),'c.','markersize',2)

set(gcf(),'position',[100 50 600 500])

legend('output','input','setpoint',1);

�ízení s kategorickým modelem

• model °ízené mince s pam¥tí

• °ízení na jednom intervalu

Pouºitý kategorický model

[ut, yt−1] yt = 1 yt = 2
1, 1 Θ1|11 Θ2|11
1, 2 Θ1|12 Θ2|12
2, 1 Θ1|21 Θ2|21
2, 2 Θ1|22 Θ2|22

P°edpoklady: Pevné a známé parametry modelu, °ízení na jednom intervalu.

Sci zna£ení: om penalizace, f = E[fp|d(t)] - expectation over y(t+ 1) fs = minf → u(t+ 1) -
optimal u

Úloha: �ízení s kategorickým modelem se známými parametry.

Poznámka

Realizuje se optimální s obecnou penalizací jednotlivých kon�gurací veli£in vystupujících v modelu
(roz²í°ený regresní vektor).

Doporu£ené experimenty

1. Penalizace s v p°ípad¥ °ízení s diskrétním modelem volí ve tvaru matice (stejných rozm¥r·
jako u modelu) a penalizuje se kaºdý stav (tj. kombinace veli£in yt, ut, yt−1) zvlá²´. Ur-
£ete si svou vlastní preferenci pro °ízení, realizujte ji penaliza£ní tabulkou Con.Cz.om a
sledujte, jak je ve výsledku realizována.

2. Zkuste m¥nit simulovanou soustavu tak, aby v ní bylo více/mén¥ neur£itosti. Sledujte
kvalitu °ízení.

Program

// Dynamic programming with discrete model (single horizon)

// fp = omega + fs (penalty function, old criterion)

// f = E[fp|d(t)] - expectation over zt(t+1)

// fs = min f => ut(t+1) - optimal ut

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

// VARIABLES TO BE SET

Con.Cz.nh=30; // length of control interval

I_detMod=0; // 1 for detrministic model: round(th)

y0=1; // initial condition for output

// zt 1 2 ut y1 = criterion

Con.Cz.om=[0 5 // 1 1

0 5 // 1 2

5 0 // 2 1

5 0]; // 2 2

// zt 1 2 ut y1 = system model

Sim.Cz.th=[.2 .8 // 1 1

.9 .1 // 1 2

.1 .9 // 2 1

.8 .2]; // 2 2

if I_detMod==1

th=round(th); // deterministický model

end

// computed variables and initializations

fs=zeros(1,2);

om=Con.Cz.om;

nh=Con.Cz.nh;

th=Sim.Cz.th;

// CONTROL LAW COMPUTATION

for t=nh:-1:1

fp=om+ones(4,1)*fs; // penalty + reminder from last step

// expectation

f=sum((fp.*th),'c'); // expectation over zt

// minimization

if f(1)<f(3), // for zt(t-1)=1

us(t,1)=1; fs(1)=f(1); // optimal control, minimum of criterion

else

us(t,1)=2; fs(1)=f(3); // optimal control, minimum of criterion

end

if f(2)<f(4), // for zt(t-1)=2

us(t,2)=1; fs(2)=f(2); // optimal control, minimum of criterion

else

us(t,2)=2; fs(2)=f(4); // optimal control, minimum of criterion

end

end

// CONTROL APPLICATION

zt(1)=y0;

for t=1:nh

ut(t+1)=us(t,zt(t)); // optimal control

i=2*(ut(t+1)-1)+zt(t); // row in the model table

zt(t+1)=(rand(1,1,'unif')>th(i,1))+1; // simulation

end

Con.Cz.zt=zt;

Con.Cz.ut=ut;

// RESULTS

plot(1:nh+1,zt,'ro')

plot(1:nh+1,ut,'g+','markersize',14)

set(gcf(),'position',[200 40 1000 600])

set(gca(),"data_bounds",[.8 nh+.2, .8 2.2])

title('Optimal control with discrete model')

legend('output','input');

41



Odhad sm¥si se statickým ukazovátkem i komponentami1

• dvourozm¥rný výstup, bez °ízení

• simulovaná data

• inicializace odhadu - za²um¥né parametry ze simulace

• standardní odhad / odhad s pevnými kovariancemi ²um· komponent

• model ukazovátka f (ct|α) = αct

Simulují se data ze sm¥si normálních, dvourozm¥rných regresních komponent a statického uka-
zovátka

yt = θi + et, i = 1, 2, · · · , nc

kde yt je výstup v £ase t,

θi =

[
(θ1)i
(θ2)i

]
jsou parametry statických komponent s dvourozm¥rným výstupem,

nc je po£et komponent.

P°edpoklady: e ∼ N (0, r) , r konstantní.

Sci zna£ení: y - yt, θ - th, r - cv.

Úloha: Simulace a odhad s modelem sm¥si statických komponent - základní kon�gurace pro
odhad sm¥si.

Poznámka

Inicializace odhadu sm¥si komponent je velice d·leºitá. Pokud nejsou po£áte£ní centra komponent
dostate£n¥ blízko dat, jsou váhy wt prakticky nulové (hodnota distribuce normálního rozd¥lení
velmi rychle klesá se vzdáleností od st°ední hodnoty) a odhad je velmi nep°esný nebo dokonce
selhává. Proto je vºdy t°eba po£áte£nímu nastavení center (st°edních hodnot) a ²í°ce (kovarian£ní
matici) odhadovaných komponent v¥novat pat°i£nou pé£i.

Zde pro inicializaci vyuºijeme znalost �skute£ných� parametr· ze simulace. Jako po£áte£ní hod-
noty parametr· v odhadu pouºijeme skute£né hodnoty parametr·, a ty více nebo mén¥ za²umíme.

Tento trik v inicializaci, který je z praktického hlediska �nefér� pouºijeme proto, ºe nám zde jde
spí²e o testování odhadu sm¥si (p°i kterém zkou²íme, co tento odhad dokáºe), neº o skute£ný
odhad, nap°. pro reálná data, kde samoz°ejm¥ �skute£né� parametry neznáme.

Doporu£ené experimenty

1. Komponenty modelu jsou statické. Tedy kaºdá komponenta je gaussovský kope£ek se st°e-
dem daným regresními koe�cienty th a ²í°kou úm¥rnou rozptylu cv. Protoºe komponenty
jsou dvourozm¥rné, pohybujeme se v rovin¥ a st°edy jsou body. Volbou st°ed· a rozptyl·
m·ºeme volit klastry, které se p°ekrývají nebo naopak jsou prakticky disjunktní. Podle

1Tato úloha je paralelní s úlohou T71MixDyn a li²í se jen modelem ukazovátka, který je zde statický.

toho je moºno nastavit situaci, kdy klasi�kace je triviální a £ekáme, ºe v²echny rozhodnutí
budou správné nebo naopak, kdy správní klasi�kace je velmi obtíºná a jsme spokojeni i s
ur£itým procentem správných rozhodnutí.
Zkuste nastavit své vlastní soustavy a ov¥°te procento správných klasi�kací.

2. Odhad modelu sm¥si je dosti obtíºný a lze konstatovat, ºe bez n¥jaké inicializace (nalezení
st°ed· komponent poblíºe center skute£ných klastr·) nebude úsp¥²ný. V této úloze se po-
uºívají po£áte£ní parametry odvozené od skute£ných parametr· soustavy (parametry ze
simulace).
Zkuste tyto parametry nastavit ru£n¥ a sledujte jejich vliv na odhad (zejména jeho po£á-
te£ní fázi).

3. Jedna z velice úsp¥²ných moºností jak p°im¥t odhad, aby se na za£átku �chytil�, je bu¤
neodhadovat rozptyly komponent a ponechat je po£áte£ní malé nebo s jejich odhadem za£ít
aº v pr·b¥hu odhadování. Odhad/neodhad rozptyl· je moºno zvolit pomocí parametru
IstCov. Zpoºd¥né odhadování je t°eba doprogramovat: if t>100, · · · , end
Zkuste odhadovat r·znými zp·soby a porovnejte výsledky.

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né komponenty.

2. Odhad se provádí podle standardních vzorc·:

(a) výpo£et vah wt pro zm¥°ený výstup yt.

(b) P°epo£et statistik komponent a ukazovátka.

(c) Konstrukce bodových odhad· parametr·. Tady je moºnost volby:

i. pr·b¥ºné odhadování kovarian£ních matic komponent,
ii. pouºití po£áte£ních kovarian£ních matic bez pr·b¥ºného p°epo£tu.

Tato varianta je bezpe£n¥j²í. P°i pr·b¥ºném odhadu se m·ºe stát, ºe jedna kom-
ponenta p°ekryje ostatní a výsledek je daný práv¥ jen touto komponentou.
Moºná je také varianta, kdy v prvé £ásti odhadování ponecháme kovarian£ní
matice pevné, a v dal²í £ásti je jiº odhadujeme.

3. Jako výsledek se ukazují hodnoty odhadovaného ukazovátka ve srovnání se simulovaným.
Tím úloha získává charakter klasi�kace.

Kód programu
Program

// Mixture estimation - static components and pointer model

// - simulated data two-dimensional data

// - initialization by parameters from simulation + noise

// - with or withou estimation of covariances of components

// FOR NC>5 IT IS NECESSARY TO SET PARAMETERS FOR SIMULATION

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

rand('seed',0)

nd=100; // number of data

nc=3; // number of componentd

I_estCov=0; // estimation of noise covariances ? 0|1 no|yes

// simulated reg.coef.

Sim.nc=nc;

Sim.Cy(1).th=[0.9 -.5]';

Sim.Cy(2).th=[0.4 0.6]';

Sim.Cy(3).th=[-.7 -.5]';

Sim.Cy(4).th=[-.1 .8]';

Sim.Cy(5).th=[ .5 -.1]';

// simulated noise covariances

Sim.Cy(1).sd=0.1*[1 0;0,1];

Sim.Cy(2).sd=0.1*[1 0;0,1];

Sim.Cy(3).sd=0.1*[1 0;0,1];

Sim.Cy(4).sd=0.1*[1 0;0,1];

Sim.Cy(5).sd=0.1*[1 0;0,1];

// simulated pointer parameters

Sim.Cp.th=fnorm(ones(1,nc)+.1,2);

Sim.ct(1)=1; // initial poiner

// initial parameters

a=.8; // std of scattering initial parametrs

for j=1:nc // from those used in simulation

[mr,mc]=size(Sim.Cy(j).th);

Ps=[Sim.Cy(j).th;1]+[a*rand(mr,mc,'n');0]; // initial parameters

Est.Cy(j).V=Ps*Ps'; // statistics

Est.Cy(j).th=Sim.Cy(j).th+a*rand(2,1,'n'); // pt.est. of reg.coef

Est.Cy(j).sd=.1*eye(2,2); // standard deviation

end

Est.ka=ones(1,nc); // counter

Est.Cp.V=ones(1,nc); // pointer statistics

Est.Cp.th=fnorm(ones(1,nc)); // pointer parameter

w=fnorm(ones(1,nc)); // weights

// SIMULATION ==========================================================

for t=2:nd

Sim.ct(t)=sum(rand(1,1,'u')>cumsum(Sim.Cp.th))+1; // pointer

j=Sim.ct(t); // active component

y=Sim.Cy(j).th+Sim.Cy(j).sd*rand(2,1,'norm'); // output

Sim.yt(:,t)=y;

end

// ESTIMATION ==========================================================

printf(' ')

for t=2:nd

if t/10==fix(t/10), printf('.'); end

for j=1:nc

[xxx,G(j)]=GaussN(Sim.yt(:,t),Est.Cy(j).th,Est.Cy(j).sd);

// proximity

end

Lq=G-max(G);

q=exp(Lq);

ww=q'.*Est.Cp.th; w=ww/sum(ww); // generation of weights

wt(:,t)=w';

// Update of statistic

Ps=[Sim.yt(:,t)' 1]; // extended reg.vec.

for i=1:nc

Est.Cy(i).V=Est.Cy(i).V+w(i)*Ps'*Ps; // information matrix

Est.ka(i)=Est.ka(i)+w(i); // counter

Est.Cp.V(i)=Est.Cp.V(i)+w(i); // pointer statistics

//nove rozdeleni informacni matice V

Vyy=Est.Cy(i).V(1:2,1:2); // part Vyy - psi.psi'

Vy=Est.Cy(i).V($,1:2); // part Vy - psi.y

V1=Est.Cy(i).V($,$); // part V1 - y.y

Est.Cy(i).th=inv(V1+1e-8*eye(V1))*Vy; // pt.est. - reg.coef.

Est.Cy(i).tht(:,t)=Est.Cy(i).th'; // pt.est. - covar.

if I_estCov~=0

// pt.est. of noise covariance - used or not

Est.Cy(i).cv=(Vyy-Vy'*inv(V1+1e-8*eye(V1))*Vy)/Est.ka(i);

end

end

Est.Cp.th=fnorm(Est.Cp.V,2); // pt.est. of pointer parameter

[ss,Est.ct(1,t)]=max(w); // store

end

// Results

disp(Sim.Cp.th,'pt.pars_sim')

disp(Est.Cp.th,'pt.pars_est')

s=2:nd;

wr=sum(Sim.ct(s)~=Est.ct(s)');

printf('\n Wrong classifications %d from %d\n',wr,length(s))

set(scf(1),'position',[50 50 400 400])

plot(s,Sim.ct(s),'bo',s,Est.ct(s),'rx')

legend('simulated','estimated');

title 'Values of the pointer'

set(scf(2),'position',[550 50 400 600])

45



for i=1:nc

subplot(nc,1,i)

plot(Est.Cy(i).tht')

title('Parameters estimate evolution '+string(i))

end

// Mixture estimation - static components and pointer model

// - real two-dimensional data

// - initialization from expert knowledge

// initial parameters are set through simulation structures + noise

// - with or without noise covariances estimation

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

rand('seed',0)

nd=100; // number of data

nc=3; // number of components

I_estCov=0; // estimation of noise covariances ? 0|1 no|yes

// MODEL INITIALIZATION

// Here expertly based initial parameters are set

// (they are copied to estimation structure Est)

// initial regression coefficients

Sim.nc=nc;

Sim.Cy(1).th=[0.9 -.5]'; // HERE SET THE INITIAL CENTERS OF COMPONENTS

Sim.Cy(2).th=[0.4 0.6]'; // -"-

Sim.Cy(3).th=[-.7 -.5]'; // -"-

// initial (or fixed) covariances

Sim.Cy(1).sd=0.1*[1 0;0,1]; // covariances of components

Sim.Cy(2).sd=0.1*[1 0;0,1]; // for I_estCov=0; stay fixed

Sim.Cy(3).sd=0.1*[1 0;0,1]; // (they are not estimated)

// initial pointer probabilities

Sim.Cp.th=fnorm([1 1 1]+.1,2);

Sim.ct(1)=1; // initial poiner value

// initial parameters

a=.8; // std of scattering initial parametrs

for j=1:nc // from those used in simulation

Ps=[Sim.Cy(j).th;1]; // initial parameters

Est.Cy(j).V=Ps*Ps'; // statistics

Est.Cy(j).th=Sim.Cy(j).th+a*rand(2,1,'n'); // pt.est. of reg.coef

Est.Cy(j).sd=.1*eye(2,2); // standard deviation

end

Est.ka=ones(1,nc); // counter

Est.Cp.V=ones(1,nc); // pointer statistics

Est.Cp.th=fnorm(ones(1,nc)); // pointer parameter

w=fnorm(ones(1,nc)); // weights

// DATA ==============================================================

load _data/dataAuto.dat // HERE, THE DATA ARE LOADED

Sim.yt=dt([1 2],1:nd); // DATA ASSIGNMENT (time runs in rows)

// ESTIMATION ========================================================

printf(' ')

for t=2:nd

if t/10==fix(t/10), printf('.'); end

for j=1:nc

[xxx,G(j)]=GaussN(Sim.yt(:,t),Est.Cy(j).th,Est.Cy(j).sd); // likelihood

end

Lq=G-max(G);

q=exp(Lq);

ww=q'.*Est.Cp.th; w=ww/sum(ww); // generation of weights

wt(:,t)=w';

// Update of statistic

Ps=[Sim.yt(:,t)' 1]; // extended reg.vec.

for i=1:nc

Est.Cy(i).V=Est.Cy(i).V+w(i)*Ps'*Ps; // information matrix

Est.ka(i)=Est.ka(i)+w(i); // counter

Est.Cp.V(i)=Est.Cp.V(i)+w(i); // pointer statistics

//nove rozdeleni informacni matice V

Vyy=Est.Cy(i).V(1:2,1:2); // part Vyy - psi.psi'

Vy=Est.Cy(i).V($,1:2); // part Vy - psi.y

V1=Est.Cy(i).V($,$); // part V1 - y.y

Est.Cy(i).th=inv(V1+1e-8*eye(V1))*Vy; // pt.est. - reg.coef.

Est.Cy(i).tht(:,t)=Est.Cy(i).th'; // pt.est. - covar.

if I_estCov~=0

// pt.est. of noise covariance - used or not

Est.Cy(i).cv=(Vyy-Vy'*inv(V1+1e-8*eye(V1))*Vy)/Est.ka(i);

end

end

47



Est.Cp.th=fnorm(Est.Cp.V,2); // pt.est. of pointer parameter

[ss,Est.ct(1,t)]=max(w); // store

end

// Results

s=2:nd;

set(scf(2),'position',[550 50 400 600])

for i=1:nc

subplot(nc,1,i)

plot(Est.Cy(i).tht')

title('Evolution of parameter estimatess '+string(i))

end

Odhad sm¥si s dynamickým ukazovátkem a statickými kom-
ponentami2

• dvourozm¥rný výstup, bez °ízení

• simulovaná data

• inicializace odhadu - za²um¥né parametry ze simulace

• standardní odhad / odhad s pevnými kovariancemi ²um· komponent

• model ukazovátka f (ct|ct−1, α) = αct|ct−1

Simulují se data ze sm¥si normálních, dvourozm¥rných regresních komponent a statického uka-
zovátka

yt = θi + et, i = 1, 2, · · · , nc

kde yt je výstup v £ase t,

θi =

[
(θ1)i
(θ2)i

]
jsou parametry statických komponent s dvourozm¥rným výstupem,

nc je po£et komponent.

P°edpoklady: e ∼ N (0, r) , r konstantní.

Sci zna£ení: y - yt, θ - th, r - cv.

Úloha: Simulace, odhad a predikce s dynamickým modelem sm¥si statických komponent -
základní kon�gurace pro odhad sm¥si.

Poznámky

1. Dynamické ukazovátko umoº¬uje predikci aktivní komponenty, protoºe dává do souvislosti
po sob¥ jdoucí aktivity komponent.

2. Inicializace odhadu sm¥si komponent je velice d·leºitá. Pokud nejsou po£áte£ní centra kom-
ponent dostate£n¥ blízko dat, jsou váhy wt prakticky nulové (hodnota distribuce normálního
rozd¥lení velmi rychle klesá se vzdáleností od st°ední hodnoty) a odhad je velmi nep°esný
nebo dokonce selhává. Proto je vºdy t°eba po£áte£nímu nastavení center (st°edních hodnot)
a ²í°ce (kovarian£ní matici) odhadovaných komponent v¥novat pat°i£nou pé£i.

Zde pro inicializaci vyuºijeme znalost �skute£ných� parametr· ze simulace. Jako po£áte£ní
hodnoty parametr· v odhadu pouºijeme skute£né hodnoty parametr·, a ty více nebo mén¥
za²umíme.

Tento trik v inicializaci, který je z praktického hlediska �nefér� pouºijeme proto, ºe nám
zde jde spí²e o testování odhadu sm¥si (p°i kterém zkou²íme, co tento odhad dokáºe), neº
o skute£ný odhad, nap°. pro reálná data, kde samoz°ejm¥ �skute£né� parametry neznáme.

2Tato úloha je paralelní s úlohou T71MixStat a li²í se jen modelem ukazovátka, který je zde dynamický.

Doporu£ené experimenty

(Jsou stejné, jako pro statickou sm¥s. Zde je uvedeme jen zkrácen¥. Podrobn¥ji jsou v T71Mix1Stat)

1. Zkuste nastavit své vlastní soustavy a ov¥°te procento správných klasi�kací.

2. Zkuste nastavit parametry ru£n¥ a sledujte jejich vliv na odhad (zejména jeho po£áte£ní
fázi).

3. Jedna z velice úsp¥²ných moºností jak p°im¥t odhad, aby se na za£átku �chytil�, je bu¤
neodhadovat rozptyly komponent a ponechat je po£áte£ní malé nebo s jejich odhadem za£ít
aº v pr·b¥hu odhadování. Odhad/neodhad rozptyl· je moºno zvolit pomocí parametru
IstCov. Zpoºd¥né odhadování je t°eba doprogramovat: if t>100, · · · , end
Zkuste odhadovat r·znými zp·soby a porovnejte výsledky.

4. (nové) Dynamické ukazovátko znamená, ºe výb¥r aktivní komponenty závisí na minulé
aktivní komponent¥. Tedy v p°epínání aktivních komponent je ur£itý °ád, daný modelem
ukazovátka (kategorický dynamický model). P°edvolený model ukazovátka je dán rovno-
m¥rnou tabulkou Sim.Cp.th.
Zvolte sv·j vlastní model ukazovátka (nap°. deterministický) a sledujte efekt, který do
odhadu takové sm¥si p°iná²í.

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né komponenty.

2. Odhad se provádí podle standardních vzorc·:

(a) výpo£et vah Wt a wt pro zm¥°ený výstup yt.

(b) P°epo£et statistik komponent a ukazovátka.

(c) Konstrukce bodových odhad· parametr·. Tady je moºnost volby:

i. pr·b¥ºné odhadování kovarian£ních matic komponent,
ii. pouºití po£áte£ních kovarian£ních matic bez pr·b¥ºného p°epo£tu.

Tato varianta je bezpe£n¥j²í. P°i pr·b¥ºném odhadu se m·ºe stát, ºe jedna kom-
ponenta p°ekryje ostatní a výsledek je daný práv¥ jen touto komponentou.

Moºná je také varianta, kdy v prvé £ásti odhadování ponecháme kovarian£ní
matice pevné, a v dal²í £ásti je jiº odhadujeme.

3. Jako výsledek se ukazují hodnoty odhadovaného ukazovátka ve srovnání se simulovaným.
Tím úloha získává charakter klasi�kace.

Kód programu

// Mixture estimation - static components and dynamic pointer model

// - simulated two-dimensional data

// - initialization by parameters from simulation + noise

// - dynamic components

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

rand('seed',0)

nd=100; // number of data

nc=3; // number of componentd

I_estCov=0; // estimation of noise covariances ? 0|1 no|yes

// simulated reg.coef.

Sim.nc=nc;

Sim.Cy(1).th=[0.9 -.5]';

Sim.Cy(2).th=[0.4 0.6]';

Sim.Cy(3).th=[-.7 -.5]';

// noise covariances

r=.1; // amplitude of noise covariances

Sim.Cy(1).sd=r*[1 0;0,1];

Sim.Cy(2).sd=r*[1 0;0,1];

Sim.Cy(3).sd=r*[1 0;0,1];

// simulated noise covariances

Sim.Cp.th=fnorm(rand(nc,nc,'u')+.1,2);

Sim.ct(1)=1; // initial poiner

// initial parameters

a=.5; // std of scattering init.params from simulated ones

for j=1:nc // from those used in simulation

[mr,mc]=size(Sim.Cy(j).th);

Ps=[Sim.Cy(j).th;1]+[a*rand(mr,mc,'n');0]; // initial parameters

Est.Cy(j).V=Ps*Ps'; // statistics

Est.Cy(j).th=Sim.Cy(j).th+a*rand(2,1,'n'); // pt.est. of reg.coef

Est.Cy(j).sd=.1*eye(2,2); // standard deviation

end

Est.ka=ones(1,nc); // counter

Est.Cp.V=.1*ones(nc,nc); // pointer statistics

Est.Cp.th=fnorm(rand(nc,nc,'u')+.1); // pointer parameter

w=fnorm(ones(1,nc)); // weights

// SIMULATION ==========================================================

for t=2:nd

i=Sim.ct(t-1); // last active component

Sim.ct(t)=sum(rand(1,1,'u')>cumsum(Sim.Cp.th(i,:)))+1; // pointer

j=Sim.ct(t); // active component

y=Sim.Cy(j).th+Sim.Cy(j).sd*rand(2,1,'norm'); // output

Sim.yt(:,t)=y; // stor

end

// ESTIMATION ==========================================================

printf(' ')

for t=2:nd

if t/10==fix(t/10), printf('.'); end

for j=1:nc

[xxx,G(j)]=GaussN(Sim.yt(:,t),Est.Cy(j).th,Est.Cy(j).sd); // likelihood

end

Lq=G-max(G); // rough normlization

q=exp(Lq); // exponent

ww=(q*w)'.*Est.Cp.th; // matrix weights

W=ww/sum(ww); // normalization - f(c(t),c(t-1)|d(t))

w=sum(W,1); // marginalization - f(c(t)|d(t))

wt(:,t)=w'; // stor

// Update of statistic

Est.ka=Est.ka+w; // counter

Est.Cp.V=Est.Cp.V+W; // ptr.stat. update

Ps=[Sim.yt(:,t)' 1]; // extended reg.vec.

for i=1:nc

Est.Cy(i).V=Est.Cy(i).V+w(i)*Ps'*Ps; // information matrix

//nove rozdeleni informacni matice V

Vyy=Est.Cy(i).V(1:2,1:2); // part Vyy - psi.psi'

Vy=Est.Cy(i).V($,1:2); // part Vy - psi.y

V1=Est.Cy(i).V($,$); // part V1 - y.y

Est.Cy(i).th=inv(V1+1e-8*eye(V1))*Vy; // pt.est. - reg.coef.

Est.Cy(i).tht(:,t)=Est.Cy(i).th(:); // pt.est. - covar.

if I_estCov~=0

// pt.est. of noise covariance - used or not

Est.Cy(i).cv=(Vyy-Vy'*inv(V1+1e-8*eye(V1))*Vy)/Est.ka(i);

end

end

Est.Cp.th=fnorm(Est.Cp.V,2); // pt.est. of pointer parameter

[ss,Est.ct(1,t)]=max(w); // store pointer values

end

// Results

disp(Sim.Cp.th,'pt.pars_sim')

disp(Est.Cp.th,'pt.pars_est')

s=2:nd;

wr=sum(Sim.ct(s)~=Est.ct(s)');

printf('\n Wrong classifications %d from %d\n',wr,length(s))

set(scf(1),'position',[50 50 400 400])

plot(s,Sim.ct(s),'bo',s,Est.ct(s),'rx')

title 'Pointer values and their predictions'

set(scf(2),'position',[550 50 400 600])

for i=1:nc

subplot(nc,1,i)

plot(Est.Cy(i).tht')

title('Evolution of parameter estimates '+string(i))

end

P°edpov¥¤ s modelem sm¥si komponent

• dvourozm¥rný výstup, bez °ízení

• simulovaná data

• model se známými parametry

• statické komponenty

• dynamický model ukazovátka f (ct|ct−1, α) = αct|ct−1

Simulují se data ze sm¥si normálních, dvourozm¥rných regresních komponent a statického uka-
zovátka

yt = θi + et, i = 1, 2, · · · , nc

kde yt je výstup v £ase t,

θi =

[
(θ1)i
(θ2)i

]
jsou parametry statických komponent s dvourozm¥rným výstupem,

nc je po£et komponent.

Délka predikce je np. Jsme v £ase t, známe y (t− 1) a práv¥ jsme zm¥°ili yt.

P°edpov¥¤ konstruujeme podle následujícího algoritmu:

• Jeden krok

1. ∀ komponenty c = 1, 2 · · ·nc spo£t¥te mc = f
(
yt|θ̂t−1

)
tj, vektor �vzdáleností� yt od jednotlivých komponent; jeho prvky jsou jsou hodnoty
hp komponent s dosazenou hodnotou yt a bodovými odhady parametr· z £asu t− 1.

2. Ur£ete bodový odhad parametru modelu ukazovátka α̂t−1

3. Vezm¥te vektor vah wt−1 z minulého kroku.

4. Ur£ete matici vah Wt, reprezentující sdruºenou pravd¥podobnost f (ct, ct−1|d (t))
takto

Wt = (wt−1m
′) . ∗ α̂t−1

kde vektory w a m jsou sloupce - tedy násobíme sloupec krát °ádek, coº dá matici, a
násobení .∗ je násobení dvou matic prvek po prvku.

5. Váhový vektor wt, který ur£uje aktivity komponent v aktuálním £ase a s ve²kerou
dostupnou informaci (v£etn¥ yt) spo£teme marginalizací, tedy vys£ítáme p°es minulé
ukazovátko (tj. se£teme Wt ve sloupcích).

6. Predikce výstupu ŷt je

ŷt =

nc∑
c=1

wcŷc;t

kde ŷc;t jsou �oby£ejné� predikce z jednotlivých komponent.

• Více krok·

Provádí se bez dal²í m¥°ené informace (budoucí hodnoty y neznáme). Predikce ukazovátka
je dána násobením váhy wt aktuálním odhadem parametru modelu ukazovátka α̂t−1.Tedy

ŵt+k = (α̂t−1)
k
wt

a op¥t

ŷt+k =

nc∑
c=1

ŵc;t+kŷc;t

kde ŷc;t jsou predikce z komponent (protoºe komponenty jsou statické, £asový index ne-
hraje roli).

P°edpoklady: e ∼ N (0, r) , r konstantní.

Sci zna£ení: y - yt, θ - th, r - cv.

Úloha: Simulace a predikce s dynamickým modelem sm¥si statických komponent - základní
kon�gurace pro predikci se sm¥sí.

Poznámky

Uvedený program je vlastn¥ �vytaºen� z programu pro odhad parametr· sm¥si a p°edpov¥¤ bu-
doucího výstupu. Zde se vynechá p°epo£et statistik a výpo£et bodových odhad· parametr·. Co
z·stává je výpo£et vah komponent. S nimi se pak kombinují predikce z jednotlivých komponent a
dají p°edpov¥¤ celé sm¥si.

Doporu£ené experimenty

1. Nastavte svoji simulovanou soustavu (s r·zným p°ekryvem simulovaných klastr·) a sle-
dujte, jak jsou jednotlivé soustavy predikovatelné.
Pozor. V rámci statických komponent nemá smysl mluvit o predikci. Predikovatelnost
tedy má smysl sledovat jako pokrytí simulovaných klastr· predikovanými. Jde tedy spí²e
o predikci komponent neº jednotlivých bod·, generovaných komponentami.

2. Program umoº¬uje nastavit délku predikce (pomocí parametru np).
Testujte, do jaké vzdálenosti má je²t¥ cenu predikovat (samoz°ejm¥ také v závislosti na
typu simulované soustavy).

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né komponenty. K dispo-
zici jsou 3 varianty modelu ukazovátka od deterministického, p°es tém¥° deterministického
aº po model s více neur£itostí. Model ale m·ºeme volit zcela libovolný, zadám matice alS
s rozm¥ry nc× nc.

2. Odhad se provádí nestandardn¥ - po£ítají se jen váhy wt vºdy vzhledem k nam¥°eným
dat·m

(a) výpo£et vah wt pro zm¥°ený výstup yt.

3. Jako hlavní výsledek se ukazují simulované a predikované klastry.

Kód programu

// Prediction with known mixture (without estimation)

// - static components

// - dynamic pointer model

// - simulated one-dimensional data

// - known model parameters

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

np=5; // number of steps of prediction

nc=3; // number of components

nd=np+1+1000; // number of data

// INITIALIZATION of simulation --------------------------------------

thS=[-2 1 5]; // centres of components

cvS=[.1 .1 .1]*1;

select 1 // these pointr models are pre-set, you can either

// chose one or write your own model. Rows must sum to 1

case 1 then, alS=[0 1 0; 0 0 1; 1 0 0];

case 2 then, alS=[.07 .87 .06; .06 .07 .87; .92 .04 .04];

case 3 then, alS=[.8 .1 .1;.1 .2 .7;.1 .1 .8]; // parameters of the pointer

end

c(1)=1; // initial values

// SIMULATION

for t=2:nd

rnd=rand(1,1,'unif');

cus=cumsum(alS(c(t-1),:));

c(t)=sum(rnd>cus)+1; // pointer generation

th=thS(c(t));

sd=sqrt(cvS(c(t)));

y(t)=th+sd*rand(1,1,'norm'); // output generation

end

w1=ones(nc,1)/nc; // initial pointer distribution

// PREDICTION

for t=2:nd // TIME LOOP ----------------------------------------------

yt=y(t); // measured output

// proximity of yt to components

for i=1:nc

m(i)=GaussN(yt,thS(i),cvS(i)); // density of normal component

end

// weighting vector

wP=fnorm(alS'*w1); // weights for prediction

wp=(alS')^np*wP; // np-steps prediction

w=fnorm(wP.*m); // normalization

// update of statistics

Ps=[yt;1]; // ext. regression vector

yp=0;

for i=1:nc

yp=yp+wp(i)*(thS(i)+sqrt(cvS(i))*rand(1,1,'norm'));

end

ytp(t+np)=yp;

w1=w; // old weighting vector

wt(:,t)=w; // stor weighting vector

wtp(:,t+np)=wp;

end // END OF TIME LOOP ---------------------------------------------

// RESULTS

[xxx ce]=max(wtp,'r'); // estimated active components

s=(np+2):nd;

wr=sum(c(s)~=ce(s)');

tx='\nWrong class. %d from %d, i.e. %d percent\n';

printf(tx,wr,length(s),100*wr/length(s))

s=(nd-100+1):nd;

plot(s,c(s),'bo','markersize',8)

plot(s,ce(s),'r.','markersize',3)

title 'Pointer and its estimate'

legend('pointer','estimate');

set(gca(),'data_bounds',[min(s) max(s) .9 nc+.1])

set(gcf(),'figure_position',[400,100])

scf(1);

set(gcf(),'figure_position',[600,150])

title 'Histograms of data and predicted data'

subplot(211)

[f1,n1]=nan_hist(y,150,'plot');

ylabel 'data'

subplot(212)

[f2,n2]=nan_hist(ytp,150,'plot');

ylabel 'predictions'

s=551:600;

scf(2);

set(gcf(),'figure_position',[200,150])

title 'time course of data and predictions'

plot(s,y(s),'ob')

plot(s,ytp(s),'xr')

legend('data','predictions');

58



Klasi�kace s modelem sm¥si komponent

• statický model f (y|x1, x2 · · · , xn)

• diskrétní výstup

• xi spojité nebo diskrétní

• hodnoty y interpretovány jako indexy t°íd v klasi�kaci

• 2 fáze: (i) u£ení s u£itelem, (ii) testování

Simulují se regresní vektory, provede se jejich lineární kombinace a ta se diskretizuje na inter-
valech. Po°adí interval· de�nuje hodnoty y.

U£ení

Regresní vektory x rozd¥líme do t°íd podle hodnot y. Regresní vektory ze t°ídy i ozna£íme Xi.
Pro kaºdou t°ídu spo£teme pr·m¥r

θi =
1

Ni

∑
x∈Xi

x

kde Ni je po£et regresních vektor· ve t°íd¥ i a dále kovarian£ní matici

ri =
1

Ni

∑
x∈Xi

(x− θi) (x− θi)
′
.

V²echny vektory jsou sloupce. Tyto odhady odpovídají odhadu sm¥si normálních komponent,
kde aktivity komponent jsou známy (u£ení s u£itelem).

Testování

Pro testování p°edpokládáme odhadnutý (nau£ený) model. Pro m¥°ené (klasi�kované) regresní
vektory provádíme �jakoby odhad�, parametry modelu uº ale nem¥níme3

Pro zm¥°ený regresní vektor x spo£teme aproximované likelihoody jednotlivých komponent - do
komponent dosadíme x (odhady parametr· máme z fáze u£ení).

Vektor likelihood· násobíme prvek po prvku vektorem stacionárních pravd¥podobností jednot-
livých komponent (tedy vektorem normovaných £etností jednotlivých t°íd tak, jak se objevily
ve fázi u£ení).

Sou£in normujeme a získáme vektor aktuálních pravd¥podobností komponent vzhledem je zm¥-
°enému x. Tento vektor nazýváme váhovým vektorem.

Vektor x klasi�kujeme do t°ídy, ur£ené po°adím (indexem) maximálního prvku váhového vek-
toru.

P°edpoklady: Postup odpovídá odhadu se sm¥sí normálních komponent.

Sci zna£ení:

xL, yL - data pro u£ení

xT - regresní vektory pro testování

Úloha: Klasi�kace dat - základní kon�gurace pro sm¥sovou klasi�kaci.
3To, ºe klasi�kaci provádíme s pevným modelem je jen záleºitost zvyku nebo napodobení klasické klasi�kace.

Ve skute£nosti nám nic nebrání pokra£ovat v odhadu modelu a tak jej dále up°es¬ovat. Toto dal²í odhadování

je jiº odhadem bez u£itele. Správné hodnoty t°íd jiº neznáme.

Poznámka: Po£áte£ní odhad modelu s u£itelem je velmi významná metoda pro inicializaci standardního odhadu

modelu sm¥si komponent. Její motivace je zakotvena práv¥ ve sm¥sové logistické regresi.

Poznámky

Program je realizován pomocí funkcí lrLearn(yL,xL) - u£ení a lrTest(xT,Est) - testování. Pokud
bychom cht¥li vid¥t skute£né algoritmy, které úlohu realizují, museli bychom se podívat do t¥chto
funkcí. Ty jsou v adresá°i _func.

Program

Popis programu

Kód programu

// Mixture used as Logistic Regression

// - learning (estimation with known pointer)

// - testing (only pointer estimation = classification)

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nL=1000; // number of data for learning

nT=150; // number of data for testing

// Simulation --------------------------

nd=nT+nL; // length of learning + testing

x=rand(nd,2,'n'); // regression vectors

x1=[x ones(nd,1)]; // -"- extended by 1

ys=x1*[1,1,2]'+.001*rand(nd,1,'n'); // continuous output

y=zeros(nd,1); // discretization

s1=find(ys<0); y(s1)=1;

s2=find((ys>=0) & (ys<3)); y(s2)=2;

s3=find(ys>=3); y(s3)=3;

yL=y(1:nL); xL=x(1:nL,:); // data for learning

// Learning ----------------------------

Est=lrLearn(yL,xL);

tT=(1:nT)+nL; // times for testing

xT=x(tT,:); // reg.vecs. for testing

// Testing -----------------------------

yT=lrTest(xT,Est); // testing

yR=y(tT)'; // estimates of tested output

// Results -----------------------------

s=1:nT;

plot(s,yR,'bo',s,yT,'rx','markersize',8)

set(gcf(),'position',[700 100 600 400])

title 'True and estimated pointer values'

printf('Wrong %d from %d\n',sum(yR~=yT),nT)

Odhad sm¥si s dynamickým ukazovátkem a diskrétními kom-
ponentami

• diskrétní výstup - °ízená koruna s pam¥tí f (yt|ut, yt−1)

• simulovaná data

• inicializace odhadu - za²um¥né parametry ze simulace

• model ukazovátka f (ct|ct−1, α) = αct|ct−1

Simulují se data ze sm¥si kategorických komponent a dynamického ukazovátka:

- komponenta i pro i = 1, 2, · · · , nc

(0.1)

ut, yt−1 yt = 1 yt = 2
1, 1 Θi

1|11 Θi
2|11

1, 2 Θi
1|12 Θi

2|12
2, 1 Θi

1|21 Θi
2|21

2, 2 Θi
1|22 Θi

2|22

kde yt a ut jsou výstup a vstup v £ase t,

- ukazovátko

ct = 1 ct = 2 · · · ct = nc
ct−1 = 1 α1|1 α2|1 αnc1

ct−1 = 2 α1|2 α2|2 αnc2

· · · · · ·
ct−1 = nc α1|nc

α2|nc
αncnc

nc je po£et komponent.

P°edpoklady:

Sci zna£ení:

Úloha: Simulace a odhad dynamickým modelem sm¥si diskrétních komponent - základní kon-
�gurace pro odhad sm¥si s diskrétními komponentami.

Poznámky

1. Dynamické ukazovátko umoº¬uje predikci aktivní komponenty, protoºe dává do souvislosti
po sob¥ jdoucí aktivity komponent.

2. Inicializace odhadu sm¥si komponent je velice d·leºitá. Bez dobré inicializace m·ºe být
konvergence v odhadu pomalá nebo odhad selºe.

Vzorec pro váhy komponent, které p°i°azují komponentám pravd¥podobnosti aktivity, se
sestává ze t°í £ástí. Jsou to (i) vektor �vzdáleností� aktuáln¥ zm¥°eného datového vzorku od
jednotlivých komponent, (ii) model ukazovátka a (iii) minulý váhový vektor wt−1. Druhá a

t°etí poloºka jsou stejné, jako u sm¥si spojitých komponent. Vzdálenost zm¥°eného datového
vzorku od komponenty je dán hodnotou modelu s dosazeným aktuálním odhadem parametr·
a hodnotami zm¥°ených dat. Model v diskrétním p°ípad¥ je dán tabulkou (0.1). Prvky této
tabulky jsou bodové odhady parametr· a zm¥°ená data yt , ut a yt−1 ur£ují jeden prvek
této tabulky. A hodnota tohoto prvku ur£uje zji²´ovanou �vzdálenost�.

Z uvedeného plyne, ºe komponenty , které mají n¥který prvek stejný, mají od p°íslu²ného
datového vzorku stejnou vzdálenost.

Zde jako po£áte£ní hodnoty parametr· pro odhad pouºijeme skute£né hodnoty parametr·,
a ty více nebo mén¥ za²umíme.

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né komponenty.

2. Odhad se provádí podle standardních vzorc·:

(a) výpo£et vah Wt a wt pro zm¥°ený výstup yt.

(b) P°epo£et statistik komponent a ukazovátka.

(c) Konstrukce bodových odhad· parametr·.

3. Jako výsledek se ukazují hodnoty odhadovaného ukazovátka ve srovnání se simulovaným.
Tím úloha získává také charakter klasi�kace.

Kód programu

// Mixture estimation - static components and dynamic pointer model

// - simulated two-dimensional data

// - initialization by parameters from simulation + noise

// - dynamic components

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

rand('seed',0)

nd=100; // number of data

nc=3; // number of componentd

I_estCov=0; // estimation of noise covariances ? 0|1 no|yes

// simulated reg.coef.

Sim.nc=nc;

Sim.Cy(1).th=[0.9 -.5]';

Sim.Cy(2).th=[0.4 0.6]';

Sim.Cy(3).th=[-.7 -.5]';

// noise covariances

r=.1; // amplitude of noise covariances

Sim.Cy(1).sd=r*[1 0;0,1];

Sim.Cy(2).sd=r*[1 0;0,1];

Sim.Cy(3).sd=r*[1 0;0,1];

// simulated noise covariances

Sim.Cp.th=fnorm(rand(nc,nc,'u')+.1,2);

Sim.ct(1)=1; // initial poiner

// initial parameters

a=.5; // std of scattering init.params from simulated ones

for j=1:nc // from those used in simulation

[mr,mc]=size(Sim.Cy(j).th);

Ps=[Sim.Cy(j).th;1]+[a*rand(mr,mc,'n');0]; // initial parameters

Est.Cy(j).V=Ps*Ps'; // statistics

Est.Cy(j).th=Sim.Cy(j).th+a*rand(2,1,'n'); // pt.est. of reg.coef

Est.Cy(j).sd=.1*eye(2,2); // standard deviation

end

Est.ka=ones(1,nc); // counter

Est.Cp.V=.1*ones(nc,nc); // pointer statistics

Est.Cp.th=fnorm(rand(nc,nc,'u')+.1); // pointer parameter

w=fnorm(ones(1,nc)); // weights

// SIMULATION ==========================================================

for t=2:nd

i=Sim.ct(t-1); // last active component

Sim.ct(t)=sum(rand(1,1,'u')>cumsum(Sim.Cp.th(i,:)))+1; // pointer

j=Sim.ct(t); // active component

y=Sim.Cy(j).th+Sim.Cy(j).sd*rand(2,1,'norm'); // output

Sim.yt(:,t)=y; // stor

end

// ESTIMATION ==========================================================

printf(' ')

for t=2:nd

if t/10==fix(t/10), printf('.'); end

for j=1:nc

[xxx,G(j)]=GaussN(Sim.yt(:,t),Est.Cy(j).th,Est.Cy(j).sd); // likelihood

end

Lq=G-max(G); // rough normlization

q=exp(Lq); // exponent

ww=(q*w)'.*Est.Cp.th; // matrix weights

W=ww/sum(ww); // normalization - f(c(t),c(t-1)|d(t))

w=sum(W,1); // marginalization - f(c(t)|d(t))

wt(:,t)=w'; // stor

// Update of statistic

Est.ka=Est.ka+w; // counter

Est.Cp.V=Est.Cp.V+W; // ptr.stat. update

Ps=[Sim.yt(:,t)' 1]; // extended reg.vec.

for i=1:nc

Est.Cy(i).V=Est.Cy(i).V+w(i)*Ps'*Ps; // information matrix

//nove rozdeleni informacni matice V

Vyy=Est.Cy(i).V(1:2,1:2); // part Vyy - psi.psi'

Vy=Est.Cy(i).V($,1:2); // part Vy - psi.y

V1=Est.Cy(i).V($,$); // part V1 - y.y

Est.Cy(i).th=inv(V1+1e-8*eye(V1))*Vy; // pt.est. - reg.coef.

Est.Cy(i).tht(:,t)=Est.Cy(i).th(:); // pt.est. - covar.

if I_estCov~=0

// pt.est. of noise covariance - used or not

Est.Cy(i).cv=(Vyy-Vy'*inv(V1+1e-8*eye(V1))*Vy)/Est.ka(i);

end

end

Est.Cp.th=fnorm(Est.Cp.V,2); // pt.est. of pointer parameter

[ss,Est.ct(1,t)]=max(w); // store pointer values

end

// Results

disp(Sim.Cp.th,'pt.pars_sim')

disp(Est.Cp.th,'pt.pars_est')

s=2:nd;

wr=sum(Sim.ct(s)~=Est.ct(s)');

printf('\n Wrong classifications %d from %d\n',wr,length(s))

set(scf(1),'position',[50 50 400 400])

plot(s,Sim.ct(s),'bo',s,Est.ct(s),'rx')

title 'Pointer values and their predictions'

set(scf(2),'position',[550 50 400 600])

for i=1:nc

subplot(nc,1,i)

plot(Est.Cy(i).tht')

title('Evolution of parameter estimates '+string(i))

end

Odhad sm¥si s dynamickým ukazovátkem a stavovými kom-
ponentami

• skalární výstup, dvourozm¥rný vstup a stav

• simulovaná data

• kovarian£ní matice stavového modelu nastaveny podle simulace

• dynamický model ukazovátka

Simulují se data ze sm¥si stavových komponent a dynamického ukazovátka:

- komponenta i pro i = 1, 2, · · · , nc

[
x1
x2

]
t

= M i

[
x1
x2

]
t−1

+N i

[
u1
u2

]
t

+ F i +

[
w1

w2

]
t

yt = Ai

[
x1
x2

]
t

+Bi

[
u1
u2

]
t

+Gi + vt

kde M i, N i, F i, Ai, Bi a Gi jsou matice stavového modelu i-té komponenty

- ukazovátko

ct = 1 ct = 2 · · · ct = nc
ct−1 = 1 α1|1 α2|1 αnc1

ct−1 = 2 α1|2 α2|2 αnc2

· · · · · ·
ct−1 = nc α1|nc

α2|nc
αncnc

nc je po£et komponent.

P°edpoklady:

Sci zna£ení:

Úloha: Simulace a odhad dynamickým modelem sm¥si stavových komponent - základní kon�-
gurace pro odhad sm¥si se stavovými komponentami.

Poznámky

1. Dynamické ukazovátko umoº¬uje predikci aktivní komponenty, protoºe dává do souvislosti
po sob¥ jdoucí aktivity komponent.

2. Vzorec pro váhy komponent, které p°i°azují komponentám pravd¥podobnosti aktivity, se
p°edchozích typech sm¥sí sestával ze t°í £ástí. Jsou to (i) vektor �vzdáleností� aktuáln¥
zm¥°eného datového vzorku od jednotlivých komponent, (ii) model ukazovátka a (iii) minulý
váhový vektor wt−1. Druhá a t°etí poloºka jsou stejné, jako u sm¥si spojitých komponent.
Vzdálenost zm¥°eného datového vzorku od komponenty se po£ítá op¥t jako hodnota datové
predikce s touto komponentou s dosazenými aktuálními daty a odhadem stavu. Tato predikce
je

f (yt|xt−1) =

ˆ
x∗t

f (yt, xt|xt−1) dxt =

ˆ
x∗t

f (yt|xt) f (xt|xt−1) dxt.

Tuto hodnotu pro kaºdou komponentu dostaneme p°ímo z Kalmanova �ltru.

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né komponenty.

2. Odhad se provádí podle standardních vzorc·:

(a) p°epo£et Kalmanových �ltr· pro kaºdou komponentu.

(b) výpo£et vah Wt a wt pro zm¥°ený výstup yt.

(c) spojení výsledk· pro jednotlivé komponenty do jednoho modelu.

3. Jako výsledek se ukazují hodnoty odhadovaného ukazovátka ve srovnání se simulovaným,
simulované a odhadnuté datová klastry a vývoj datové predikce.

Kód programu

// Mixture estimation with state-space components

//

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=150; // number of data

nc=3; // number of components

// init of simulatiom

// component 1

Sim.Cy(1).M=[.3 -.3; // reg.coef for simulation

.5 .1];

Sim.Cy(1).N=[.2 -.3

.3 .5];

Sim.Cy(1).A=[.6 .4];

Sim.Cy(1).B=[0.3 0.7];

Sim.Cy(1).F=[-3 -7]';

// component 2

Sim.Cy(2).M=[.05 -.3;

.4 -.1];

Sim.Cy(2).N=[.2 -.3

.3 .5];

Sim.Cy(2).A=[.4 0.8];

Sim.Cy(2).B=[-0.4 -0.5];

Sim.Cy(2).F=[0 0]';

// component 3

Sim.Cy(3).M=[.6 -.1;

.8 -.5];

Sim.Cy(3).N=[.2 -.3

.3 .5];

Sim.Cy(3).A=[1 -.4];

Sim.Cy(3).B=[0.4 0.5];

Sim.Cy(3).F=[3 .4]';

Sim.rv=1e-3; // common coves of output model

Sim.rw=1e-3; // common coves of state model

ct=zeros(1,nd);

ct(1)=1; // pointer initial condition

Sim.Cp.th=[1 0 0;

0 1 0;

0 0 1]+1;

Sim.Cp.th=fnorm(Sim.Cp.th,2); // pointer model parameter

nx=max(size(Sim.Cy(1).M));

[ny,nu]=size(Sim.Cy(1).B);

ut=rand(nu,nd,'n');

yt=zeros(ny,nd);

xt=zeros(nx,nd);

x=zeros(nx,1);

// SIMULATION ==========================================================

for t=2:nd

ct(t)=sum(rand(1,1,'u')>cumsum(Sim.Cp.th(ct(t-1),:)))+1;

j=ct(t); // index of active component

u=ut(:,t);

x=Sim.Cy(j).M*x+Sim.Cy(j).N*u+Sim.Cy(j).F+sqrt(Sim.rw)*rand(nx,1,'n');

y=Sim.Cy(j).A*x+Sim.Cy(j).B*u+sqrt(Sim.rv)*rand(ny,1,'n');

yt(:,t)=y; xt(:,t)=x;

end

Sim.xt=xt;

Sim.yt=yt;

Sim.ut=ut;

// init of estimation

Est.Cp.V=1e-1*ones(nc,nc); // pointer statistics

Est.Cp.th=fnorm(Est.Cp.V,2); // pointer parameter

w1=zeros(nc,1); w1(1)=1; // initial ptr.value

sx=zeros(nx,1); // initial state

rx=1e3*eye(nx,nx); // initial covariance matrix

// of the state estimate

//ESTIMATION ===========================================================

printf(' '), tt=fix(nd/10);

for t=(2:nd)

if t/tt==fix(t/tt), printf('.'); end

for j=1:nc

[Est.Cy(j).x,Est.Cy(j).rx,yp,ry]=.. // Kalman filter

KalmanXY(sx,yt(:,t),ut(:,t),Sim.Cy(j).M,Sim.Cy(j).N,..

Sim.Cy(j).A,Sim.Cy(j).B,Sim.Cy(j).F,Sim.rw,Sim.rv,rx);

[xxx,Lq(j)]=GaussN(yt(:,t),yp,ry); // likelihood

Est.Cy(j).yp(t)=yp;

end

Lqq=Lq-max(Lq); // rough normalization

q=exp(Lqq); // exponent from logarithm

Wp=(w1*q').*Est.Cp.th; W=Wp/sum(Wp); // matrix weight for pointer

wp=sum(W,'r')'; w=wp/sum(wp); // weights for components

// updated components are merged with the weights w

sx=0; s1=0; s2=0;

for i=1:nc // KL approximation of expectation

sx=sx+w(i)*Est.Cy(i).x;

end

xe(:,t)=sx;

for i=1:nc // KL approximation of covariance

s1=s1+w(i)*(Est.Cy(i).x-sx)*(Est.Cy(i).x-sx)';

s2=s2+w(i)*Est.Cy(i).rx;

end

rx=s1+s2;

Est.Cp.V=Est.Cp.V+W;

Est.Cp.th=fnorm(nu,2);

w1=w;

wt(:,t)=w;

end

Est.xt=xe;

Est.wt=wt;

Est.rx=rx;

// RESULTS

s=15:nd;

[xxx ce]=max(wt,'r'); wr=sum(ct(s)~=ce(s));

printf('\nWrong classifications %d from %d\n',wr,length(s))

set(scf(1),'position',[70 70 500 500])

plot(1:nd,ct,'o',1:nd,ce,'.')

title('Simulated and estimated pointer values')

set(scf(2),'position',[700 70 500 500])

plot(xt(1,s),xt(2,s),'bx','markersize',10)

plot(xe(1,s),xe(2,s),'ro')

title('Simulated and estimated clusters')

col=['b','r','g','k','m'];

set(scf(3),'position',[300 270 500 500])

title('Output predictions')

for i=1:nc

plot(Est.Cy(i).yp,col(i))

end

legend('first','second','third');

Odhad sm¥si s dynamickým ukazovátkem a statickými kom-
ponentami4

• smí²ený (spojitý i diskrétní) jednorozm¥rný výstup, bez °ízení

• simulovaná data

• inicializace odhadu - datový vzorek klasi�kovaný expertem

• standardní odhad / odhad s pevnými kovariancemi ²um· komponent

• zna£ení:

� yt spojitý výstup

� zt diskrétní výstup

� ct pointer

• modely:

� model spojité komponenty

f (yt|yt−1, zt, zt−1,Θc)

� model diskrétní komponenty

f (zt|zt−1, β) = βzt|zt−1

� model ukazovátka
f (ct|ct−1, zt−1, α) = αct|ct−1;zt−1

P°edpoklady: diskrétní modely závisí jen na diskrétních veli£inách.

Sci zna£ení: standard podle dohody o zna£ení.

Úloha: Simulace, odhad a predikce s dynamickým modelem sm¥si statických komponent. Úloha
pat°í do nadstavby.

Poznámky

Program

Popis programu

1. Simulace hodnot ukazovátka a podle toho generování dat z p°íslu²né spojité a diskrétní
komponenty.

2. Inicializace odhadovacího algoritmu se provádí s mnoºinou apriorních dat. Tato mnoºina se
sestává z datových záznam· (bu¤ p°edem zm¥°ených nebo um¥le zkonstruovaných tak, aby
dob°e reprezentovali datový prostor). K t¥mto datovým záznam·m jsou expertn¥ p°i°azeny
hodnoty ukazovátka (t°ídy, do kterých pat°í). Je n¥kolik moºností, jak tuto p°ed-klasi�kaci
získat:

4Tato úloha je zobecn¥ním jak spojitých tak i diskrétních sm¥si. Pat°í uº mezi nadstandardn¥ sloºité záleºitosti

(a) pro získání dat pouºijeme speciální m¥°ení (nap°. délky kolon jsou zaznamenávány
studenty),

(b) expert sleduje m¥°ení a ur£uje, ko které kategorie pat°í (nap°. stupe¬ dopravy)

(c) expert sestavuje datové záznamy um¥le, tak aby odpovídaly jeho p°edstavám o fun-
gování systému (nap°. tato zatá£ka (s parametry ...) pat°í do kategorie velmi nebez-
pe£ných, a tahle (s parametry ...) je bez nebezpe£í - p°itom zatá£ky nemusí v·bec
existovat)

Tato inicializace je velmi ú£inná a ve standardním p°ípad¥ sta£í n¥jakých 5 dat do kaºdé
komponent a algoritmus je velmi dob°e p°ipraven

3. Odhad se provádí podle modi�kovaných vzorc· ze spojitého a diskrétního odhadu:

(a) výpo£et vah Wt a wt pro zm¥°ený výstup yt.

(b) P°epo£et statistik komponent a ukazovátka.

(c) Konstrukce bodových odhad· parametr·. Tady je moºnost volby:

i. pr·b¥ºné odhadování kovarian£ních matic komponent,
ii. pouºití po£áte£ních kovarian£ních matic bez pr·b¥ºného p°epo£tu.

Tato varianta je bezpe£n¥j²í. P°i pr·b¥ºném odhadu se m·ºe stát, ºe jedna kom-
ponenta p°ekryje ostatní a výsledek je daný práv¥ jen touto komponentou.
Moºná je také varianta, kdy v prvé £ásti odhadování ponecháme kovarian£ní
matice pevné, a v dal²í £ásti je jiº odhadujeme.

4. Jako výsledek se ukazují hodnoty odhadovaného ukazovátka ve srovnání se simulovaným.
Tím úloha získává charakter klasi�kace.

Kód programu

// Estimation of mixture with DATA DEPENDENT DYNAMIC POINTER

// Models:

// Cy - continuous componen f( yt(t) | yt(t-1),zt(t),zt(t-1),theta )

// Cz - discrete component f( zt(t) | zt(t-1),be )

// Cp - discrete pointer model f( ct(t) | ct(t-1),zt(t-1),als )

// ct=1,2,3; zt=1,2;

// - mixed (discrete and continuous) modeled data

// - pre-classified data sample for initialization

[u,t,n]=file(); // find working directory

chdir(dirname(n(1))); // set working directory

clear("u","t","n") // clear auxiliary data

exec("ScIntro.sce",-1),mode(0) // intro to sesion

nd=150; // number of data

ni=20; // number of initial data

nc=3; // number of components

// parameters for continuous components

Sim.Cy(1).th=[.95 1 1 5];

Sim.Cy(2).th=[.6 -1 1 0];

Sim.Cy(3).th=[.81 1 -1 -5];

// common variance for continuous components

r=.1;

// parameters for discrete components

Sim.Cz(1).th=fnorm([1 0

0 1]+1,2);

Sim.Cz(2).th=fnorm([0 1

1 0]+2,2);

Sim.Cz(3).th=fnorm([1 0

0 1]+3,2);

// parameters for pointer model

Sim.Ca(1).th=fnorm([0 1 0

0 0 1

1 0 0]+5,2);

Sim.Ca(2).th=fnorm([0 0 1

1 0 0

0 1 0]+8,2);

// initial conditions

yt(1)=0; zt(1)=1; ct(1)=1;

// SIMULATION ---------------------------------------------------------

for t=2:nd

als=Sim.Ca(zt(t-1)).th; // parameter of Cp

ct(t)=sum(rand(1,1,'unif')>cumsum(als(ct(t-1),:)))+1; // pointer value

be=Sim.Cz(ct(t)).th; // parameter of Cz

zt(t)=sum(rand(1,1,'unif')>cumsum(be(zt(t-1),:)))+1; // disc. output

ps=[yt(t-1) zt(t) zt(t-1) 1]; // regression vector of Cy

th=Sim.Cy(ct(t)).th; // parameters of Cy

yt(t)=th*ps'+sqrt(r)*rand(1,1,'norm'); // cont. output

end // ----------------------------------------------------------------

ky=min(size(yt)); // dimension of cont. output

mz=max(zt); // number of values of disc. output

// Initial statistics and par. estimates

for j=1:nc

// model Cy

Est.Cy(j).V=zeros(5,5);

Est.Cy(j).cv=r;

// model Cz

Est.Cz(j).V=rand(mz,mz,'u')+.1*ones(mz,mz);

Est.Cz(j).th=fnorm(Est.Cz(j).V);

end

74



// initial estimates of th = estimation with known components !!!

// (for initiation, a sample of pre-classified data is used)

for t=2:ni

Ps=[yt(t) yt(t-1) zt(t) zt(t-1) 1]';

Est.Cy(ct(t)).V=Est.Cy(ct(t)).V+Ps*Ps';

Vy=Est.Cy(ct(t)).V(1,1); // part Vy yt.yt

Vyps=Est.Cy(ct(t)).V(2:$,1); // part Vyps psi.yt

Vps=Est.Cy(ct(t)).V(2:$,2:$); // part Vps psi.psi'

Est.Cy(ct(t)).th=inv(Vps+1e-5*eye(Vps))*Vyps; // pt.est. of reg.coef.

end

// poiner

Est.ka=5*ones(1,nc);

for j=1:mz

// statistics for pointer

Est.Cp(j).V=rand(nc,nc,'u')+.1*ones(nc,nc);

// parameters of pointer model

Est.Cp(j).th=fnorm(Est.Cp(j).V,2);

end

// weighting vector

w=fnorm(rand(1,nc,'u')+ones(1,nc));

// TIME LOOP ==========================================================

printf(' '), tt=fix(nd/10);

for t=(2:nd)

if t/tt==fix(t/tt), printf('.'); end

// computation of likelihoods

ps=[yt(t-1) zt(t) zt(t-1) 1]'; // regression vector

for j=1:nc // likelihood for yt

yp=Est.Cy(j).th'*ps;

rh=Est.Cy(j).cv;

[xxx Lm(j)]=GaussN(yt(t),yp,rh);

Zp(j)=Est.Cz(j).th(zt(t-1),zt(t)); // prediction for zt

end

Lm=Lm-max(Lm);

Yp=exp(Lm);

// computation of weights W and w

al=Est.Cp(zt(t-1)).th;

Wp=((Yp.*Zp)*w)'.*al;

if sum(Wp)<1e-6, Wp=rand(nc,nc,'u'); end

W=Wp/sum(Wp); // matrix W

w=sum(W,'r'); // weighting vector w

wt(:,t)=w';

[xxx ce(t)]=max(w);

// update of statistics

Ps=[yt(t) yt(t-1) zt(t) zt(t-1) 1]'; // extended regression vector

for j=1:nc // update of Cy.V, w, Cz.V

Est.Cy(j).V=Est.Cy(j).V+w(j)*Ps*Ps';

Est.ka(j)=Est.ka(j)+w(j);

Est.Cz(j).V(zt(t-1),zt(t))=Est.Cz(j).V(zt(t-1),zt(t))+w(j);

end

Est.Cp(zt(t-1)).V= Est.Cp(zt(t-1)).V+W; // update of Ca

// computation of point estimates

for j=1:nc

Vy=Est.Cy(j).V(1,1); // part Vy

Vyps=Est.Cy(j).V(2:$,1); // part Vyps

Vps=Est.Cy(j).V(2:$,2:$); // part Vps

Est.Cy(j).th=inv(Vps+1e-5*eye(Vps))*Vyps; //regression coefs

th(j).t(:,t)=Est.Cy(j).th;

// how about estimation of coveriances:

// for 0 - fix variance, for 1 - variances estimation

if 1

Est.Cy(j).r=(Vy-Vyps'*inv(Vps+1e-5*eye(Vps))*Vyps)/Est.ka(j);

end

Est.Cz(j).th=fnorm(Est.Cz(j).V,2); // parameters for Cz

end

Est.Cp(zt(t-1)).th=fnorm(Est.Cp(zt(t-1)).V,2); // pointer pt.est.

end // END OF TIME LOOP ===============================================

// Results

s=2:nd;

wr=sum(ct(s)~=ce(s));

printf('\nWrong classifications %d from %d\n',wr,length(s))

set(scf(1),'position',[600 50 600 600])

plot(1:nd,ct','x',1:nd,ce','.r')

set(gca(),'data_bounds',[1 nd .8 nc+.2])

legend('simulation','estimation'); // 5 means place legend hy hand

xlabel('Time [periods]')

ylabel('Pointer values')

title('The pointer estimation')

set(scf(4),'position',[150 50 400 500])

for i=1:nc

subplot(nc,1,i)

plot(th(i).t')

xlabel('Time [periods]')

ylabel('Parameter values')

title('Evolution of parameter estimation of the '+..

string(i)+'. normal component')

end

